

# TI-99/4A: 51 Fun and Educational Programs

GIL M. SCHECHTER



**TI 99/4A:  
51 Fun and Educational  
Programs**



Photo by Judi Edwards

**Gil M. Schechter** has over 20 years experience in the computer field. He has held various computer-related positions with the Martin Company and Westinghouse Electric Company, working in the areas of computer hardware and software design and artificial intelligence. He has been employed by Londontown Corporation since 1970, working initially as an EDP systems analyst and, more recently, in the application of computers to forecasting and marketing analysis. Mr. Schechter currently holds the position of Director of Marketing Research.

Mr. Schechter received a degree in electrical engineering from the University of Florida in 1963. He has also attended George Washington University and Johns Hopkins. Mr. Schechter is currently active in teaching several microcomputer courses to both children and adults.

# **TI 99/4A: 51 Fun and Educational Programs**

**by  
Gil M. Schechter**

**Howard W. Sams & Co., Inc.**  
4300 WEST 62ND ST. INDIANAPOLIS, INDIANA 46268 USA

Copyright © 1983 by Gil M. Schechter

FIRST EDITION  
FIFTH PRINTING — 1984

All rights reserved. No part of this book shall be reproduced, stored in a retrieval system, or transmitted by any means, electronic, mechanical, photocopying, recording, or otherwise, without written permission from the publisher. No patent liability is assumed with respect to the use of the information contained herein. While every precaution has been taken in the preparation of this book, the publisher assumes no responsibility for errors or omissions. Neither is any liability assumed for damages resulting from the use of the information contained herein.

International Standard Book Number: 0-672-22192-6  
Library of Congress Catalog Card Number: 83-50373

Edited by *Frank N. Speights*  
Illustrated by *Jill E. Martin*

*Printed in the United States of America.*

# PREFACE

If you have a Texas Instruments TI 99/4A computer, and are looking for programs that are *fun* and *educational*, then you want this book.

It contains four major sections that let you use the book in many ways:

1. It will provide hours of *fun and entertainment* with its many puzzles, games, and graphics.
2. It will *educate* with programs such as Spelling Test, State Capitals, Fractions-to-Decimals, Language Translator, and Planets.
3. It provides you with *working examples* of TI BASIC in action. Whether you are just learning BASIC or are looking for some new ideas or tricks, much can be gained by examining these programs and trying to understand how they work.
4. The programs in the book can be *easily changed and expanded*. New words can be added to word games. The time limit can be changed in the game of "BOMB." The vocabulary of Language Translator can be updated.

I wrote this book with the entire family in mind — from pre-school children who will be fascinated by games, such as TRAMPOLINE and TWINKLE, TWINKLE, LITTLE STAR, to the older members of the family who may benefit from learning more complex subjects, such as the understanding of binary numbers and SPEED READING. With 51 programs, there should be something of interest for everyone. You can use these programs as they are or you can change them to fit your own special needs.

Most of all, I hope these 51 programs make your TI 99/4A computer much more valuable and much more fun to use.

GIL M. SCHECHTER

*This book is dedicated to Carole, Jeff, and Lori.*

# CONTENTS

## PART I

LEARNING COMPUTING BY WORKING WITH NUMBERS AND CHARACTERS .....	9
Fill the screen with your name .....	11
What's your batting average? .....	11
How far away is the lightning? .....	12
If I double my investment every year, how much will it be worth? ..	13
A routine to round off decimal numbers .....	13
A program that can print backwards .....	14
Changing from fractions to decimals .....	15
A routine to right justify numbers .....	16
A program to sort in alphabetical order .....	17
Con-Cat-E-Nation .....	18
Let the computer balance your checkbook .....	19
Let the computer figure your family budget .....	21
Turn your computer into a tv typewriter .....	21
A program to help you understand random numbers .....	22
A program to help you understand binary numbers .....	23
A program to help you understand hexadecimal numbers .....	24
Create and test your own special characters .....	26
Creating a bar graph .....	28

## PART II

HAVING FUN WITH GRAPHICS AND SOUNDS .....	31
The Face .....	33
Rocket .....	34
Twinkle, Twinkle, Little Star .....	34
Siren .....	35
Joystick Crayon .....	35
Keyboard Crayon .....	37
Computer Piano .....	38



Missile Control . . . . .	40
Name That Tone . . . . .	42
Trampoline . . . . .	44

### **PART III**

<b>HAVING FUN WITH PUZZLES AND GAMES . . . . .</b>	<b>47</b>
Horse Race . . . . .	49
Bomb . . . . .	51
Remove the X's . . . . .	53
Repeat . . . . .	55
Scramble . . . . .	56
Sorting Batting Averages . . . . .	58
Dice . . . . .	59
Paper-Rock-Scissors . . . . .	61
Secret Code . . . . .	62
Micro Hangman . . . . .	64
Zodiac Signs . . . . .	66

### **PART IV**

<b>USING EDUCATIONAL PROGRAMS FOR EASIER LEARNING . . . . .</b>	<b>69</b>
Spelling Test . . . . .	71
Coin Count . . . . .	72
Checking Math Homework . . . . .	74
Guess the Missing Letters . . . . .	75
Savings Account . . . . .	77
Planets . . . . .	78
Temperatures . . . . .	79
Speed Reading . . . . .	81
State Capitals . . . . .	82
English to Metric . . . . .	84
Area of a Rectangle . . . . .	85
Language Translator . . . . .	86

# PART I

## **Learning Computing by Working With Numbers and Characters**



## Fill the screen with your name

If you like your name, you will *love* this program. The computer will ask you to "ENTER ANY NAME". Enter your own name, or that of a friend, and the name that you just entered will fill the screen.

```
100 REM -REPEAT NAME
110 CALL CLEAR
120 PRINT "*** REPEAT-A-NAME ***"
130 PRINT
140 PRINT
150 PRINT "ENTER ANY NAME"
160 INPUT NAME$
170 CALL CLEAR
180 FOR I=1 TO 100
190 PRINT NAME$;" ";
200 NEXT I
210 PRINT
220 PRINT
230 PRINT "DO YOU WISH TO DO THIS AGAIN? (YES OR NO)"
240 INPUT DECISION$
250 IF DECISION$="YES" THEN 110
260 IF DECISION$="NO" THEN 300
270 PRINT
280 PRINT "PLEASE ANSWER YES OR NO"
290 GOTO 210
300 END
```

## What's your batting average?

To find out your average, just enter the number of times you were at bat and the number of hits you had. The computer will figure out your batting average and display it on the screen. You can find the batting average for as many players as you like. Try it for some of your favorite players in the major leagues.

When you are finished and want to end the program, hit the FUNCTION-CLEAR keys.

```
100 REM -BATTING AVERAGE
110 CALL CLEAR
120 PRINT "BATTING AVERAGE"
130 GOSUB 250
140 INPUT "ENTER AT-BAT,HITS ":A,H
150 IF H<=A THEN 190
160 PRINT
170 PRINT "*** ERROR-MORE HITS THAN AT-BATS"
180 GOTO 130
190 AVG=INT(H/A*1000)
200 GOSUB 250
210 PRINT "YOU HAD ";H;" HITS"
```

```

220 PRINT "YOU CAME TO BAT ";A;"TIMES"
230 PRINT "YOUR BATTING AVERAGE IS ";AVG
240 GOTO 130
250 REM SKIP 3 LINES
260 FOR I=1 TO 3
270 PRINT
280 NEXT I
290 RETURN

```

## How far away is the lightning?

Have you ever been in a thunderstorm and noticed the number of seconds that pass from the time you see the lightning and hear the resulting thunder? This difference in time can tell you approximately how far away the lightning is. That's because you see the light almost instantly, whereas the sound travels much more slowly.

Use this program to set up a table of distances for every second from one to twenty (1, 2, 3, . . . , 20 seconds). Then, the next time you are in a thunderstorm, note the time between the lightning and thunder and use your table to find the distance between you and the lightning.

```

100 REM -LIGHTNING
110 CALL CLEAR
120 PRINT "***** LIGHTNING *****"
130 PRINT :::
140 PRINT "THIS PROGRAM CALCULATES THE ":"DISTANCE
    BETWEEN YOU AND":"THE LIGHTNING."
150 PRINT :::"WHEN YOU SEE THE LIGHTNING ":"FLASH, NOTE
    THE NUMBER OF"
160 PRINT "SECONDS UNTIL YOU HEAR THE THUNDER."
170 PRINT :::"ENTER NUMBER OF SECONDS"
180 INPUT S
190 CALL CLEAR
200 FT=1087*S
210 MI=FT/5280
220 MET=331*S
230 KM=MET/1000
240 PRINT "THE DISTANCE BETWEEN YOU":::"AND THE
    LIGHTNING":::"IS APPROXIMATELY:"
250 PRINT :::
260 PRINT FT;" FEET"
270 PRINT
280 PRINT MI;" MILES"
290 PRINT
300 PRINT MET;" METERS"
310 PRINT
320 PRINT KM;" KILOMETERS"
330 PRINT :::
340 END

```

## **If I double my investment every year, how much will it be worth?**

Wouldn't it be nice to own a stock, mutual fund, or other investment that doubles in value every year. If you could do this, even a small amount of money would become quite large in just a few years.

This program lets you experiment with different starting amounts and tells you how much your investment is worth at the end of every year, for seven years. So try it for yourself. Start with small investments, such as \$100, \$300, \$1000, etc., and see how quickly your money can grow.

```
100 REM -DOUBLE
110 CALL CLEAR
120 PRINT "*** DOUBLE YOUR MONEY ***"
130 PRINT
140 PRINT
150 PRINT "SEE HOW QUICKLY YOUR MONEY GROWS IF YOU
    DOUBLE IT 7 YEARS IN A ROW"
160 PRINT
170 PRINT
180 PRINT "ENTER STARTING DOLLARS "
190 INPUT S
200 GOSUB 240
210 FOR Y=1 TO 7
220 S=S*2
230 PRINT Y,S
240 NEXT Y
250 GOTO 340
260 REM HEADING
270 CALL CLEAR
280 PRINT "WATCH YOUR MONEY GROW"
290 PRINT
300 PRINT
310 PRINT "YEAR", "DOLLARS"
320 PRINT
330 RETURN
340 END
```

## **A routine to round off decimal numbers**

Here is a simple method you can use to round off a number. This program rounds numbers to the nearest hundredth. For example, enter 15.7356 and the program will round it to 15.74. Or, enter 1.234 and the program will round it off to 1.23.

The rounding routine is given in lines 200 and 210. It is very handy and you may want to use it in some of your programs in the future. It works like this:

Start with our example number	15.7356
Add	0.0050
	<hr/>
	15.7406
 Multiply by 100	 1574.06
Take the integer value	1574
Divide by 100	15.74

Had our number been 15.7346, adding the value 0.0050 would have given 15.7396, which would have rounded to 15.73.

```

100 REM -ROUNDING A NUMBER
110 CALL CLEAR
120 PRINT
130 PRINT "*** ROUNDING A NUMBER ***"
140 PRINT
150 PRINT
160 PRINT "THIS PROGRAM SHOWS HOW TO ROUND A NUMBER TO
    THE NEAREST HUNDREDTH."
170 PRINT
180 PRINT
190 INPUT "ENTER NUMBER TO BE ROUNDED (SUCH AS
    15.7356):";N
200 N=100*(N+.005)
210 N=INT(N)/100
220 PRINT
230 PRINT
240 PRINT N
250 PRINT
260 PRINT
270 GOTO 190

```

## A program that can print backwards

It's nice to work with characters and strings, because you can make the computer do some interesting things.

Computers normally print forward. This little program changes that and lets the computer print backwards.

Just enter your word, and it will be reversed.

```

100 REM PRINT IN REVERSE
110 CALL CLEAR
120 PRINT "**** PRINT IN REVERSE ****"
130 GOSUB 350
140 GOSUB 350
150 PRINT "ENTER WORDS TO BE REVERSED (TO END, TYPE
    END)"
160 INPUT W$
170 IF W$="END" THEN 400
180 CALL CLEAR

```

```

190 BW$=" "
200 L=LEN(W$)
210 FOR I=L TO 1 STEP -1
220 L$=SEG$(W$,I,1)
230 BW$=BW$&L$
240 NEXT I
250 GOSUB 350
260 PRINT W$
270 PRINT
280 PRINT "-----"
290 PRINT
300 PRINT BW$
310 GOSUB 350
320 GOSUB 350
330 GOSUB 350
340 GOTO 150
350 REM -SKIP 3 LINES
360 FOR J=1 TO 3
370 PRINT
380 NEXT J
390 RETURN
400 END

```

## Changing from fractions to decimals

You can use this program to change a number from its fractional form to its decimal form.

Enter the fraction as you would write it (for example, 1/5 or 2/3). This is somewhat different from most such programs, which require that you enter the numerator, a comma, and then the denominator. That's because this program is able to recognize the numerator as the number before the slash (/), and the denominator as the number after the slash (/).

To end the program, press the FUNCTION-CLEAR keys.

```

100 REM -FRACTION TO DECIMAL
110 CALL CLEAR
120 PRINT "*** FRACTION TO DECIMAL ***"
130 PRINT
140 PRINT
150 NUM$=" "
160 DENOM$=" "
170 SWITCH=0
180 PRINT
190 PRINT
200 PRINT "ENTER FRACTION (SUCH AS 1/5) "
210 PRINT
220 INPUT X$
230 L=LEN(X$)
240 FOR I=1 TO L
250 N$=SEG$(X$,I,1)
260 IF SWITCH=1 THEN 300
270 IF N$="/" THEN 360
280 NUM$=NUM$&N$

```



```

290 GOTO 310
300 DENOM$=DENOM$&N$
310 NEXT I
320 DECIMAL=VAL(NUM$)/VAL(DENOM$)
330 PRINT
340 PRINT X$;"=";DECIMAL
350 GOTO 150
360 REM SWITCH
370 SWITCH=1
380 GOTO 310

```

## A routine to right justify numbers

Your TI computer normally prints numbers in a left justified way. This means that numbers of different lengths are lined up on the left side. For example:

```

1234
 4
275

```

However, numbers are more readable when they are right justified, as shown below:

```

1234
 4
275

```

The routine to right justify numbers begins in line 290 and can handle any number up to seven digits. You can use it in your own programs to make your numbers more readable.

```

100 REM -"RIGHT JUSTIFY"
110 REM -PROGRAM CONTAINS
120 REM -A ROUTINE TO RIGHT
130 REM -JUSTIFY A NUMBER
140 REM -BEFORE PRINTING IT.
150 REM -NUMBERS ARE
160 REM -JUSTIFIED TO 7
170 REM -POSITIONS
180 CALL CLEAR
190 PRINT "**** RIGHT JUSTIFY ****"
200 PRINT
210 PRINT
220 READ A$,B
230 IF A$="END" THEN 430
240 X=B
250 GOSUB 290
260 B$=X$
270 PRINT A$,B$
280 GOTO 220
290 REM =RIGHT JUSTIFY
300 X$=STR$(X)

```

```

310 L=LEN(X$)
320 IF L>=7 THEN 350
330 X$=" "&X$
340 GOTO 310
350 RETURN
360 DATA JONES,1234
370 DATA HARRIS,4
380 DATA JOHNSON,12345
390 DATA FRANKLIN,280
400 DATA KING,456678
410 DATA KLINE,1234567
420 DATA END,0
430 END

```

## A program to sort in alphabetical order

You can use this program to sort a list of names into alphabetical order. Just place the names that are to be sorted in the DATA statements. Following the last name in the last DATA statement, insert an END to let the program know where the list ends. (Don't confuse this with the END statement given in line 510, which designates the end of the program.)

```

100 REM -"ALPHABETIC ORDER"
110 CALL CLEAR
120 PRINT "*** ALPHABETICAL ORDER ***"
130 PRINT
140 PRINT
150 PRINT "SORTS NAMES IN ALPHABETICAL"
160 PRINT "ORDER"
170 PRINT
180 PRINT "LAST NAME TO BE SORTED"
190 PRINT "MUST BE END"
200 PRINT
210 PRINT "ALL NAMES ARE IN DATA STATEMENTS"
220 PRINT
230 PRINT
240 PRINT "HIT ENTER TO SORT"
250 INPUT ENTER$
260 DIM X$(50)
270 Z=1
280 READ A$
290 IF A$="END" THEN 330
300 X$(Z)=A$
310 Z=Z+1
320 GOTO 280
330 GOSUB 390
340 CALL CLEAR
350 FOR I=1 TO Z
360 PRINT X$(I)
370 NEXT I
380 GOTO 510
390 REM =ALPHABET SORT
400 FOR I=1 TO Z
410 FOR J=I+1 TO Z
420 IF X$(I)<X$(J) THEN 460

```

```

430 H$=X$(I)
440 X$(I)=X$(J)
450 X$(J)=H$
460 NEXT J
470 NEXT I
480 RETURN
490 DATA DAN, IAN, PAUL, SUE, FRED
500 DATA PETE, JOE, SAM, HARRY, STEVE, END
510 END

```

## Con-Cat-E-Nation

Concatenation means "linking" or "putting together." This program illustrates concatenation by letting you create words from a group of letters and syllables.

RUN the program. You will see nine letters and syllables. How many words can you create from these? For instance, to make the word "STAND," concatenate letter number 8, letter number 5, and syllable number 7. Enter these as a single number, 857.

You can easily change the nine letters and syllables given in the program by replacing them with other characters in the DATA statements. Up to nine letters or syllables may be used and they must be followed by the word END.

```

100 REM -CONCATENATE THE SYLLABLES
110 REM -UP TO 9 SYLLABLES MAY BE PLACED IN DATA
    STATEMENT. LAST SYLLABLE MUST BE END.
120 REM -MAIN
130 GOSUB 220
140 GOSUB 300
150 GOSUB 400
160 GOSUB 510
170 PRINT "TO CONTINUE, HIT ENTER."
180 PRINT "TO END, TYPE END."
190 INPUT RESPONSE$
200 IF RESPONSE$="END" THEN 570
210 GOTO 140
220 REM -READ SYLLABLES
230 FOR I=1 TO 9
240 READ LET$
250 IF LET$="END" THEN 280
260 SYL$(I)=LET$
270 NEXT I
280 Z=I-1
290 RETURN
300 REM -PRINT SCREEN
310 CALL CLEAR
320 PRINT "-CONCATENATE THE SYLLABLES-"
330 GOSUB 510
340 FOR K=1 TO Z

```

```

350 PRINT K;SYL$(K)
360 NEXT K
370 GOSUB 510
380 PRINT "CREATE WORDS BY ENTERING THE NUMBER OF THE
      SYLLABLES YOU WISH TO CONCATENATE      (EXAMPLE:
      857)"
390 RETURN
400 REM -CREATE AND PRINT
410 INPUT NUM$
420 L=LEN(NUM$)
430 WORD$=""
440 FOR I=1 TO L
450 X=VAL(SEG$(NUM$,I,1))
460 WORD$=WORD$&SYL$(X)
470 NEXT I
480 GOSUB 510
490 PRINT "YOUR WORD IS: ";WORD$
500 RETURN
510 REM -SKIP 3 LINES
520 FOR J=1 TO 3
530 PRINT
540 NEXT J
550 RETURN
560 DATA IN,ER,G,R,T,E,AND,S,CA,END
570 END

```

## Let the computer balance your checkbook

Here is a useful program that will help you balance your family checking account. To understand how it works, RUN the program using the following example:

STARTING BALANCE = \$500  
DEPOSITS = \$200

The computer will show that:

SUM OF CHECKS = \$434.39  
NEW BALANCE = \$265.61

The outstanding checks are contained in the DATA statements. Each DATA statement contains the check number and the dollar amount of the check. To use this program for your own checking account, replace the example checks given in the program with your own outstanding checks.

After you have changed the DATA statements to list your own checks, save the program for use in the following month.

```

100 REM CHECKBOOK
110 CALL CLEAR
120 PRINT "**** CHECKBOOK ****"
130 PRINT
140 PRINT
150 PRINT "ENTER STARTING BALANCE"
160 INPUT BAL
170 PRINT
180 PRINT
190 PRINT "ENTER DEPOSITS"
200 INPUT DEP
210 READ CHKNO$,AMT
220 IF CHKNO$="END" THEN 250
230 CHKS=CHKS+AMT
240 GOTO 210
250 CALL CLEAR
260 PRINT "STARTING BALANCE=";BAL
270 PRINT
280 PRINT "DEPOSITS=";DEP
290 PRINT
300 PRINT "SUM OF CHECKS=";CHKS
310 PRINT
320 PRINT "NEW BAL=";BAL+DEP-CHKS
330 PRINT
340 PRINT
350 RESTORE
360 PRINT "CHECK NO","AMOUNT"
370 PRINT
380 READ CHKNO$,AMT
390 IF CHKNO$="END" THEN 480
400 PRINT CHKNO$,AMT
410 N=N+1
420 IF N<6 THEN 470
430 PRINT
440 PRINT "HIT ENTER TO CONTINUE"
450 INPUT ENTER$
460 N=0
470 GOTO 380
480 PRINT " ", "-----"
490 PRINT "TOTAL",CHKS
500 PRINT
510 PRINT
520 DATA 100,24.73
530 DATA 101,14.21
540 DATA 102,16.21
550 DATA 103,24.73
560 DATA 108,27.04
570 DATA 110,11.43
580 DATA 111,12.99
590 DATA 112,47.23
600 DATA 114,22.05
610 DATA 115,77.12
620 DATA 118,10.05
630 DATA 119,33.39
640 DATA 121,17.98
650 DATA 128,36.65
660 DATA 130,44.09
670 DATA 133,14.49
680 DATA END,0
690 END

```

## Let the computer figure your family budget

Your home computer can also be used to figure your family budget. Each budget item (such as food, car, gasoline, etc.) is contained in the DATA statements. To use the program for your own budget, replace the example DATA statements given in the program with DATA statements that reflect your own situation.

RUN the program to get the total budget and then save it for future use. Later, if any budget item changes, just retrieve the program, make whatever changes are necessary by using your EDIT functions, and RUN the program to get your new total. Again, save the changed program.

```
100 REM  FAMILY BUDGET
110 CALL CLEAR
120 PRINT "*** FAMILY BUDGET ***"
130 PRINT
140 PRINT
150 PRINT "ITEM", "AMOUNT"
160 PRINT
170 PRINT
180 READ ITEM$, AMT
190 IF ITEM$="END" THEN 230
200 PRINT ITEM$, AMT
210 TOTAL=TOTAL+AMT
220 GOTO 180
230 PRINT " ", "-----"
240 PRINT "TOTAL", TOTAL
250 PRINT
260 PRINT
270 END
280 DATA FOOD,100
290 DATA RENT,95
300 DATA GAS,25
310 DATA ELECTRICITY,15
320 DATA PHONE,4
330 DATA CAR,50
340 DATA CLOTHING,20
350 DATA SAVINGS,20
360 DATA END,0
```

## Turn your computer into a tv typewriter

Here is a way to practice your typing without using paper or a typewriter. This program lets you type directly on the screen. Begin typing when you see the message "PLEASE BEGIN TYPING."

To end the program, hit the FUNCTION-CLEAR keys.

```

100 REM -TYPEWRITER
110 REM
120 REM -THIS PROGRAM SHOWS
130 REM -HOW CALL KEY MAY BE USED
140 REM -TO TYPE DIRECTLY
150 REM -ON THE SCREEN
160 CALL CLEAR
170 PRINT "*** PLEASE BEGIN TYPING ***"
180 PRINT
190 PRINT
200 CALL KEY(0,KEY,STATUS)
210 IF KEY<32 THEN 200
220 IF STATUS<1 THEN 200
230 PRINT CHR$(KEY);
240 GOTO 200

```

## A program to help you understand random numbers

The purpose of this program is to help you understand how random numbers are generated and how they work. In this example, we will limit ourselves to generating only three numbers: 1, 2, or 3. Since these numbers are being generated *randomly*, it is pure luck or chance as to whether we will get a 1, 2, or 3. For example, if we generate 300 numbers, we should get about the same number of 1s, 2s, and 3s (in this case, 100 ones, 100 twos, and 100 threes).

Try it. Start with 100 numbers. Then, try 200, 300, and 400 numbers. The amount of ones, twos, and threes should be close, if not exactly the same. In fact, the more numbers you generate, the more equal should be the amount of ones, twos, and threes that you get, but it will take longer to get the answer.

While the numbers are being generated, the message, "PLEASE WAIT, I'M THINKING", will appear on the screen.

```

100 CALL CLEAR
110 REM -"  RANDOM NUMBERS"
120 PRINT "*** RANDOM NUMBERS ***"
130 PRINT
140 PRINT
150 PRINT "DEMONSTRATION OF RANDOM"
160 PRINT "NUMBER GENERATOR."
170 PRINT
180 PRINT "RND WILL GENERATE"
190 PRINT "EITHER 1, 2, OR 3. PROGRAM"
200 PRINT "WILL COUNT HOW MANY TIMES"
210 PRINT "EACH OCCURS. YOU ENTER"
220 PRINT "HOW MANY NUMBERS "
230 PRINT "YOU WANT TO GENERATE."

```

```

240 PRINT
250 PRINT
260 PRINT "HIT ENTER TO START"
270 PRINT
280 PRINT
290 INPUT ENTER$
300 CALL CLEAR
310 INPUT "HOW MANY NUMBERS DO YOU WANT TO GENERATE ":N
320 CALL CLEAR
330 PRINT "PLEASE WAIT, I'M THINKING"
340 FOR I=1 TO N
350 RANDOMIZE
360 R=INT(RND*3)+1
370 ON R GOSUB 400,420,440
380 NEXT I
390 GOTO 460
400 N1=N1+1
410 RETURN
420 N2=N2+1
430 RETURN
440 N3=N3+1
450 RETURN
460 CALL CLEAR
470 PRINT "NUMBER 1 OCCURRED=";N1
480 PRINT "NUMBER 2 OCCURRED=";N2
490 PRINT "NUMBER 3 OCCURRED=";N3
500 PRINT
510 PRINT "NUMBERS GENERATED=";N1+N2+N3
520 END

```

## A program to help you understand binary numbers

An understanding of binary numbers is very important for anyone who is seriously interested in knowing how computers work. This program will help you get a better understanding of binary numbers and will show you how to convert from binary numbers to decimal numbers.

Run the program. Enter the suggested binary number 1101, which is equal to 13 in decimal. The computer will show you how it calculated 13 decimal from 1101 binary.

Repeat with other binary numbers, such as 111, 101, 1000, etc. After a little practice, you should become quite familiar with the binary numbering system.

```

100 REM BINARY NUMBERS
110 CALL CLEAR
120 PRINT "*** BINARY NUMBERS ***"
130 PRINT
140 PRINT
150 PRINT
160 PRINT
170 PRINT
180 PRINT

```



```

190 N=0
200 T=0
210 PRINT "ENTER BINARY NUMBER MADE UP OF ONES AND
    ZEROS (SUCH AS 1101). TO END, TYPE END."
220 INPUT B$
230 IF B$="END" THEN 450
240 PRINT
250 PRINT "-----"
260 L=LEN(B$)
270 FOR I=L TO 1 STEP -1
280 BIT$=SEG$(B$,I,1)
290 BIT=VAL(BIT$)
300 IF BIT>1 THEN 360
310 PRINT BIT;" X 2 TO POWER OF";N;"=" ";BIT*(2^N)
320 T=T+(BIT*2^N)
330 N=N+1
340 NEXT I
350 GOTO 390
360 PRINT
370 PRINT "***ERROR-NUMBER MUST CONTAIN ONLY ONES OR
    ZEROS."
380 GOTO 150
390 REM -PRINT ANSWER
400 PRINT
410 PRINT
420 PRINT B$;"=";T
430 PRINT "-----"
440 GOTO 150
450 END

```

## A program to help you understand hexadecimal numbers

Hexadecimal numbers are a little more complicated than binary numbers, but they are just as important for a thorough understanding of how computers work. This program should help you get such an understanding, by letting you experiment with the changing of hexadecimal numbers into decimal numbers.

Whereas binary numbers use a base of 2, hexadecimal numbers use a base of 16. Many computer books deal with the hexadecimal numbering system, so if you are not familiar with it, it is suggested that you obtain such a book and review the subject first.

When you first run this program, start by inputting the suggested hexadecimal number FF (which is equal to 255 in decimal). Then, try some other hexadecimal numbers, such as A2, C5, F5, etc. The computer will show you how each hex number is converted into a decimal number.

```


100 REM  HEXADECIMAL NUMBERS
110 CALL CLEAR
120 PRINT "*** HEXADECIMAL NUMBERS ***"
130 PRINT
140 PRINT
150 PRINT
160 PRINT
170 PRINT
180 PRINT
190 N=0
200 T=0
210 PRINT "ENTER HEXADECIMAL NUMBER      (SUCH AS FF) "
220 INPUT H$
230 PRINT
240 PRINT "-----"
250 L=LEN(H$)
260 FOR I=L TO 1 STEP -1
270 HEX$=SEG$(H$,I,1)
280 IF HEX$>"F" THEN 420
290 IF HEX$="A" THEN 520
300 IF HEX$="B" THEN 540
310 IF HEX$="C" THEN 560
320 IF HEX$="D" THEN 580
330 IF HEX$="E" THEN 600
340 IF HEX$="F" THEN 620
350 HEX=VAL(HEX$)
360 IF HEX>15 THEN 420
370 PRINT HEX;" X 16 ^";N;"=" ";HEX*(16^N)
380 T=T+(HEX*16^N)
390 N=N+1
400 NEXT I
410 GOTO 450
420 PRINT
430 PRINT "*** ERROR-HEX NUMBERS MUST  CONTAIN 0-9,
      OR A-F"
440 GOTO 150
450 REM -PRINT ANSWER
460 PRINT
470 PRINT
480 PRINT H$;"=";T
490 PRINT "-----"
500 GOTO 150
510 REM  DETERMINE DECIMAL VALUE
520 HEX=10
530 GOTO 370
540 HEX=11
550 GOTO 370
560 HEX=12
570 GOTO 370
580 HEX=13
590 GOTO 370
600 HEX=14
610 GOTO 370
620 HEX=15
630 GOTO 370

```

## Create and test your own special characters

In TI BASIC, you can create your own special graphics characters by using the CALL CHAR subprogram. This program helps you develop and test those special graphics characters. In order to use CALL CHAR, however, you must supply a hexadecimal pattern identifier, which represents your special character. (See pages II-76 to II-79 of your User's Reference Guide.) This program makes it easier to develop this hexadecimal identifier and to test it.

For example, suppose you wanted to use this program to help you create and test a special character that looks like the one in Fig. 1-1.

	ENTER THIS PATTERN OF ONES AND ZEROS TO DESCRIBE EACH ROW	THIS PROGRAM WILL GIVE YOU THE HEXADECIMAL IDENTIFIER CODE
	1 1 1 1 1 1 1 1 1 0 0 0 0 0 0 1 1 0 0 0 0 0 0 1 1 0 0 1 1 0 0 1 1 0 0 1 1 0 0 1 1 0 0 0 0 0 0 1 1 0 0 0 0 0 0 1 1 1 1 1 1 1 1 1	FF 81 81 99 99 81 81 FF

Here are the steps to follow:

1. Run the program
2. To create the hexadecimal pattern identifier code, enter the word CODE.
3. Enter a pattern of ones and zeros to describe each row (a blank square is represented by a zero while a shaded square is represented by a one). The pattern for each row of our character is shown to the right of the sketch in Fig. 1-1.
4. For each pattern of 8 ones and zeros which describe a row, this program will calculate the hex code. For example, the hex code for the first row is FF (which corresponds to a pattern of 11111111).

5. When you finish all eight rows you should have the hex code, FF 81 81 99 99 81 81 FF, to describe this character.
6. You are now finished generating the hex code. Type DONE and hit ENTER.
7. It's time now to test the hex code. Enter TEST.
8. To test, enter the 16-character hex code FF 81 81 99 99 81 81 FF.
9. Our special character should now appear on the screen.
10. To end the program, enter END.

```

100 REM  CREATE AND TEST SPECIAL CHARACTER
110 CALL CLEAR
120 PRINT "----- SPECIAL CHARACTERS -----"
130 PRINT
140 PRINT
150 PRINT "TO CREATE CODE: TYPE CODE"
160 PRINT
170 PRINT "TO TEST CHARACTER: TYPE TEST"
180 PRINT
190 PRINT "TO END: TYPE END"
200 INPUT CHOICE$
210 IF CHOICE$="CODE" THEN 260
220 IF CHOICE$="TEST" THEN 740
230 IF CHOICE$="END" THEN 920
240 PRINT "ERROR-PLEASE ENTER AGAIN"
250 GOTO 130
260 CALL CLEAR
270 PRINT "SPECIAL CHARACTER CODE GENERATOR"
280 PRINT
290 PRINT
300 DIM HEX$(16)
310 FOR I=0 TO 9
320 HEX$(I)=STR$(I)
330 NEXT I
340 HEX$(10)="A"
350 HEX$(11)="B"
360 HEX$(12)="C"
370 HEX$(13)="D"
380 HEX$(14)="E"
390 HEX$(15)="F"
400 PRINT
410 PRINT
420 PRINT
430 PRINT "ENTER PATTERN OF 8 ONES AND ZEROS (WHEN
    DONE: TYPE DONE)"
440 INPUT X$
450 IF X$="DONE" THEN 110
460 IF LEN(X$)<>8 THEN 400
470 REM  DECODE RIGHT 4 BITS
480 N=0
490 V=0
500 FOR I=8 TO 5 STEP -1
510 B$=SEG$(X$,I,1)
520 IF B$="0" THEN 540

```

```

530 V=V+2^N
540 N=N+1
550 NEXT I
560 V2$=HEX$(V)
570 REM DECODE LEFT 4 BITS
580 N=0
590 V=0
600 FOR I=4 TO 1 STEP -1
610 B$=SEG$(X$,I,1)
620 IF B$="0" THEN 640
630 V=V+2^N
640 N=N+1
650 NEXT I
660 V1$=HEX$(V)
670 CODE$=V1$&V2$
680 PRINT
690 PRINT "-----"
700 PRINT X$;"=";CODE$
710 PRINT
720 PRINT "-----"
730 GOTO 400
740 REM TEST CHARACTER
750 CALL CLEAR
760 PRINT "**** TEST CHARACTER ****"
770 PRINT
780 PRINT
790 PRINT "ENTER 16-CHARACTER TEST CODE (EXAMPLE:
    FF818199998181FF)"
800 INPUT PAT$
810 IF LEN(PAT$)=16 THEN 860
820 PRINT
830 PRINT "ERROR: MUST ENTER 16 CHAR"
840 PRINT
850 GOTO 790
860 CALL CHAR(140,PAT$)
870 CALL CLEAR
880 CALL VCHAR(12,17,140)
890 PRINT "WHEN DONE HIT ENTER"
900 INPUT ENTER$
910 GOTO 110
920 END

```

## Creating a bar graph

In a more serious application of the graphics capabilities of your computer, this program is designed to illustrate how data can be presented in a bar graph form.

The numbers to be graphed are shown in the DATA statements. The last such number must be -999 to indicate the end of the data.

You may easily apply this program to your own needs by using your own numbers instead of those used in the example statements. The program has the ability to scale the data so that all the numbers will fit on the screen.

```

100 REM -"BAR GRAPH"
110 CALL CLEAR
120 PRINT "***** BAR GRAPH *****"
130 PRINT
140 PRINT
150 PRINT "PROGRAM TO ILLUSTRATE"
160 PRINT "HOW BAR GRAPH CAN BE"
170 PRINT "CONSTRUCTED. VERTICAL"
180 PRINT "DATA IS IN DATA STATEMENTS."
190 PRINT
200 PRINT "ONLY POSITIVE NUMBERS CAN BE GRAPHED WITH
THIS PROGRAM."
210 PRINT
220 PRINT "HIT ENTER TO CONTINUE."
230 PRINT
240 PRINT
250 INPUT ENTER$
260 X=X+2
270 CALL CLEAR
280 B$="FFFFFFFFFFFFFFF"
290 V$="1818181818181818"
300 K$="FFFF000000000000"
310 CALL CHAR(130,B$)
320 CALL CHAR(140,V$)
330 CALL CHAR(150,K$)
340 CALL VCHAR(3,3,140,20)
350 CALL HCHAR(20,1,150,30)
360 READ Z
370 IF Z<0 THEN 410
380 IF Z<29 THEN 360
390 Z9=Z
400 GOTO 360
410 SCALE=INT(29/20+1)
420 GOSUB 520
430 X=3
440 RESTORE
450 READ Y
460 Y=INT(Y/SCALE)
470 IF Y<0 THEN 510
480 X=X+2
490 CALL VCHAR((20-Y),X,130,Y)
500 GOTO 450
510 GOTO 510
520 REM =GRAPH TITLE
530 T$="BAR GRAPH"
540 FOR I=1 TO 9
550 SS$=SEG$(T$,I,1)
560 CALL HCHAR(3,(20+I),ASC(SS$))
570 NEXT I
580 RETURN
590 DATA 40,90,80,60,100,110,120
600 DATA -999

```



## PART II

### **Having Fun With Graphics and Sounds**





## The Face

RUN this program.

No, this is not a self-portrait of the author. It's just a way to welcome you to this section on Graphics and Sounds.

Like *The Face*, the programs in this section are intended to amuse you and make you happy.

```
100 REM FACE
110 CALL CLEAR
120 BLOCK$="FFFFFFFFFFFFFFF"
130 VSIDE$="0303030303030303"
140 HSIDE$="000000000000FFFF"
150 LBLOCK$="FEFCF8F0E0C08000"
160 RBLOCK$="7F3F1F0F07030100"
170 CALL CHAR(140,BLOCK$)
180 CALL CHAR(141,VSIDE$)
190 CALL CHAR(142,HSIDE$)
200 CALL CHAR(143,LBLOCK$)
210 CALL CHAR(144,RBLOCK$)
220 READ R,C,CODE,REP
230 IF CODE=0 THEN 420
240 CALL VCHAR(R,C,CODE,REP)
250 GOTO 220
260 REM -HI
270 DATA 4,13,140,3
280 DATA 5,14,140,1
290 DATA 4,15,140,3
300 DATA 4,20,140,3
310 REM -EYES & NOSE
320 DATA 11,13,140,2
330 DATA 11,14,140,2
340 DATA 11,19,140,2
350 DATA 11,20,140,2
360 DATA 14,16,140,2
370 DATA 14,17,140,2
380 REM -VSIDES
390 DATA 9,10,141,13
400 DATA 9,23,141,13
410 DATA 0,0,0,0
420 READ R,C,CODE,REP
430 IF CODE=0 THEN 560
440 CALL HCHAR(R,C,CODE,REP)
450 GOTO 420
460 REM -TOP&BOTTOM
470 DATA 8,11,142,13
480 DATA 21,11,142,13
490 REM -MOUTH
500 DATA 16,13,144,1
510 DATA 17,14,144,1
520 DATA 16,20,143,1
530 DATA 17,19,143,1
540 DATA 18,15,140,4
550 DATA 0,0,0,0
560 GOTO 560
```

## Rocket

This very simple program uses the TAB function to create the illusion of a rocket blasting off into space. Sound effects are also provided, so be sure to turn up your volume.

```
100 REM -"ROCKET"
110 CALL CLEAR
120 PRINT "***** ROCKET *****"
130 PRINT
140 PRINT
150 PRINT "A ROCKET IS ABOUT TO BLAST"
160 PRINT "OFF. TO START THE COUNTDOWN"
170 PRINT "HIT ENTER."
180 INPUT E$
190 CALL CLEAR
200 FOR I=10 TO 0 STEP -1
210 CALL SOUND(800,800,2)
220 PRINT I
230 NEXT I
240 CALL CLEAR
250 PRINT TAB(17);"*"
260 PRINT TAB(16);"* *"
270 PRINT TAB(15);"*  *"
280 FOR I=1 TO 9
290 PRINT TAB(14);"*    *"
300 NEXT I
310 PRINT TAB(14);"*****"
320 FOR I=1 TO 25
330 PRINT
340 CALL SOUND(500,-7,27-I,150,27-I)
350 NEXT I
360 FOR V=2 TO 30 STEP 4
370 CALL SOUND(400,-7,V,150,V)
380 NEXT V
390 END
```

## Twinkle, Twinkle, Little Star

RUN this program to hear a computerized rendition of an old and favorite song.

```
100 REM -"TWINKLING STARS"
110 CALL CLEAR
120 READ F
130 IF F=0 THEN 230
140 CALL SOUND(400,F,2)
150 C$=C$&"*"
160 PRINT C$
170 GOTO 120
180 DATA 698,698,1047,1047
190 DATA 1175,1175,1047,1047
200 DATA 932,932,880,880
210 DATA 784,784,698
220 DATA 0
```

```

230 FOR I=1 TO 250
240 NEXT I
250 RESTORE
260 N=N+1
270 IF N=4 THEN 290
280 GOTO 120
290 CALL CLEAR
300 PRINT TAB(10);"THE END"
310 FOR I=1 TO 10
320 PRINT
330 NEXT I
340 END

```

## Siren

This program produces a siren sound. It is also an excellent example of how the CALL SOUND statement can be combined with the FOR-NEXT loop.

```

100 REM -SIREN
110 CALL CLEAR
120 PRINT "***** SIREN *****"
130 PRINT
140 PRINT
150 PRINT "HIT ENTER TO START SIREN"
160 INPUT ENTER$
170 PRINT
180 PRINT
190 PRINT "TO STOP: HIT FUNCTION AND 4-KEY TOGETHER!"
200 FOR F=500 TO 800 STEP 20
210 CALL SOUND(-100,F,2)
220 NEXT F
230 FOR F=800 TO 500 STEP -20
240 CALL SOUND(-100,F,2)
250 NEXT F
260 GOTO 200

```

## Joystick Crayon

Here is your chance to be a computer artist. This program makes it possible for you to draw pictures on the screen with one of your joysticks.

Just follow these easy steps:

1. Make sure ALPHA LOCK (in the lower left-hand corner of your keyboard) is in the UP position.
2. RUN the program. A tiny black square should appear on the screen.
3. Test both joysticks to see which one moves the tiny black square. Use this joystick.

4. Move the joystick in the direction you wish to draw.
5. To move the black square without drawing a line or to erase an existing line, hold down the FIRE button (red button on joystick). With this button down, moving the black square over an existing line will erase that part of the line.

If you don't have joysticks, use the *Keyboard Crayon* program given next.

```

100 REM -JOYSTICK CRAYON
110 CALL CLEAR
120 PRINT "***** JOYSTICK CRAYON *****"
130 PRINT
140 PRINT
150 PRINT "USE ONE OF YOUR JOYSTICKS"
160 PRINT "TO DRAW LINES ON THE SCREEN."
170 PRINT
180 PRINT "HOLD THE FIRE BUTTON DOWN TO"
190 PRINT "ERASE OR MOVE WITHOUT DRAWING A LINE."
200 PRINT
210 PRINT "BEFORE STARTING, MAKE SURE"
220 PRINT "ALPHA-LOCK BUTTON ON KEYBOARD IS UP."
230 PRINT
240 PRINT
250 PRINT "HIT ENTER TO START"
260 INPUT ENTER$
270 CALL CLEAR
280 W$="FFFFFFFFFFFFFFF"
290 CALL CHAR(140,W$)
300 R=12
310 C=16
320 CALL JOYST(1,X,Y)
330 CALL KEY(1,KEY,STAT)
340 IF KEY(>)18 THEN 360
350 CALL HCHAR(R,C,32)
360 IF X>0 THEN 470
370 IF Y>0 THEN 420
380 IF Y<0 THEN 520
390 IF X<0 THEN 570
400 CALL VCHAR(R,C,140)
410 GOTO 320
420 REM -JOY UP
430 R=R-1
440 IF R>1 THEN 460
450 R=1
460 GOTO 400
470 REM -JOY RIGHT
480 C=C+1
490 IF C<32 THEN 510
500 C=32
510 GOTO 400
520 REM -JOY DOWN
530 R=R+1
540 IF R<24 THEN 560
550 R=24

```

```

560 GOTO 400
570 REM -JOY LEFT
580 C=C-1
590 IF C>1 THEN 610
600 C=1
610 GOTO 400
620 END

```

## Keyboard Crayon

Even if you don't own joysticks, you can still be a computer artist. This program is similar to the preceding program, *Joystick Crayon*, except that it lets you draw on the screen by using your keyboard, as follows:

Key	What Key Will Do
E	DRAW UP
X	DRAW DOWN
D	DRAW RIGHT
S	DRAW LEFT
B	ERASE OR MOVE SQUARE WITHOUT DRAWING A LINE (BLANK MODE).
C	CLEAR THE SCREEN
W	BEGIN WRITING AGAIN AFTER BLANK MODE (WRITE MODE)

Note: Keys E, X, D, and S draw in the same direction as the arrows on the keys.

Unlike the *Joystick Crayon* program, ALPHA LOCK need not be UP for this program. To end the program, hit the FUNCTION-CLEAR keys.

```

100 REM KEYBOARD CRAYON
110 CALL CLEAR
120 PRINT "**** KEYBOARD CRAYON ****"
130 PRINT
140 PRINT "USE YOUR KEYBOARD"
150 PRINT "TO DRAW LINES ON THE SCREEN."
160 PRINT
170 PRINT "PRESS E TO DRAW UP"
180 PRINT "PRESS X TO DRAW DOWN"
190 PRINT
200 PRINT "PRESS D TO DRAW RIGHT"
210 PRINT "PRESS S TO DRAW LEFT"
220 PRINT
230 PRINT "TO ERASE OR MOVE WITHOUT DRAWING A LINE ON  
THE SCREEN -- HIT B (BLANK MODE)"
240 PRINT
250 PRINT "TO START WRITING AGAIN, HIT W (WRITE MODE)"

```

```

260 PRINT
270 PRINT "TO CLEAR THE SCREEN AT ANY TIME --- HIT C
(CLEAR)"
280 PRINT
290 PRINT
300 PRINT "HIT ENTER TO START"
310 INPUT ENTER$
320 CALL CLEAR
330 W$="FFFFFFFFFFFFFFF"
340 CALL CHAR(140,W$)
350 R=12
360 C=16
370 WRITE=1
380 CALL KEY(0,KEY,STAT)
390 IF KEY(>)67 THEN 410
400 CALL CLEAR
410 IF KEY(>)66 THEN 430
420 WRITE=0
430 IF KEY(>)87 THEN 450
440 WRITE=1
450 IF WRITE=1 THEN 470
460 CALL HCHAR(R,C,32)
470 IF KEY=68 THEN 580
480 IF KEY=69 THEN 530
490 IF KEY=88 THEN 630
500 IF KEY=83 THEN 680
510 CALL VCHAR(R,C,140)
520 GOTO 380
530 REM -UP
540 R=R-1
550 IF R>1 THEN 570
560 R=1
570 GOTO 510
580 REM -RIGHT
590 C=C+1
600 IF C<32 THEN 620
610 C=32
620 GOTO 510
630 REM -DOWN
640 R=R+1
650 IF R<24 THEN 670
660 R=24
670 GOTO 510
680 REM -LEFT
690 C=C-1
700 IF C>1 THEN 720
710 C=1
720 GOTO 510
730 END

```

## Computer Piano

Now, it's time to bring out the musical genius within you. RUN this program and your computer will become a musical instrument. If you hit the A-key, it will play an A note. Likewise, the B-key will play a B note, and so on for notes A through G.

The sharp notes are played as follows:

Note	Press Key
A sharp	H
C sharp	I
D sharp	J
F sharp	K
G sharp	L

Eventually you will want to compose your own music. But meanwhile, to get you started, here are two classical pieces:

### **Mary Had a Little Lamb**

EDCD EEE  
DDD  
EGG  
EDCD EEE  
EDD EDC

### **Jingle Bells**

EEE  
EEE  
EG CDE  
FFF  
FFEE  
EGG  
EDC

```
100 CALL CLEAR
110 GOSUB 550
120 CALL KEY(0,N,S)
130 IF N<65 THEN 120
140 IF N>76 THEN 120
150 IF N=65 THEN 270
160 IF N=66 THEN 310
170 IF N=67 THEN 330
180 IF N=68 THEN 370
190 IF N=69 THEN 410
200 IF N=70 THEN 430
210 IF N=71 THEN 470
220 IF N=72 THEN 290
230 IF N=73 THEN 350
240 IF N=74 THEN 390
250 IF N=75 THEN 450
260 IF N=76 THEN 490
270 F=220
280 GOTO 510
290 F=233
300 GOTO 510
310 F=247
```



```

320 GOTO 510
330 F=262
340 GOTO 510
350 F=277
360 GOTO 510
370 F=294
380 GOTO 510
390 F=311
400 GOTO 510
410 F=330
420 GOTO 510
430 F=349
440 GOTO 510
450 F=370
460 GOTO 510
470 F=392
480 GOTO 510
490 F=415
500 GOTO 510
510 CALL SOUND(100,F,2)
520 CALL KEY(0,X,S)
530 IF X<>N THEN 120
540 GOTO 510
550 REM INSTRUCTIONS
560 PRINT "*** COMPUTER PIANO ***"
570 PRINT
580 PRINT
590 PRINT "PRESS KEY FOR NOTE"
600 PRINT
610 PRINT "NOTE", "SHARP"
620 PRINT
630 PRINT "A", "H"
640 PRINT "B", " "
650 PRINT "C", "I"
660 PRINT "D", "J"
670 PRINT
680 PRINT "E", " "
690 PRINT "F", "K"
700 PRINT "G", "L"
710 PRINT
720 PRINT
730 RETURN

```

## Missile Control

You are sitting at the controls of a battery of SAM missiles (surface-to-air missiles). You hear that the enemy is about to launch ten bombing attacks. If your aim is good, you can destroy the enemy's bombers and help win the war.

The enemy will start his attack when you press the ENTER key. Fire your missile whenever you think the path of the missile will cross the path of the bomber (causing a "hit" to occur). To fire your missile, press the "F" key.

After each attack and at the end of the game, the computer will give you your score.

```

100 REM -MISSILE CONTROL
110 CALL CLEAR
120 PRINT
130 PRINT "ALERT! ALERT!"
140 PRINT
150 PRINT
160 PRINT "ENEMY BOMBERS ARE ABOUT"
170 PRINT "TO MAKE 10 BOMBING RUNS"
180 PRINT
190 PRINT "YOU CAN STOP THEM BY LAUNCHING SAM MISSILES"
200 PRINT
210 PRINT "PRESS THE F KEY TO FIRE THE MISSILES"
220 PRINT
230 PRINT
240 PRINT "HIT ENTER TO START"
250 INPUT ENTER$
260 CALL CLEAR
270 FIRE=0
280 RAIDS=RAIDS+1
290 GOSUB 630
300 GOSUB 360
310 CALL KEY(0,KEY,STAT)
320 IF KEY(>)70 THEN 340
330 FIRE=1
340 GOSUB 430
350 GOTO 300
360 REM -MOVE PLANE
370 CALL HCHAR(RP,CP,32,2)
380 CP=CP+1
390 IF CP<32 THEN 410
400 GOTO 790
410 CALL HCHAR(RP,CP,140,2)
420 RETURN
430 REM -MOVE ROCKET
440 IF FIRE=0 THEN 530
450 CALL VCHAR(RR,CR,32)
460 RR=RR-1
470 IF RR<2 THEN 790
480 CALL VCHAR(RR,CR,150)
490 CALL GCHAR(RR-1,CR,X)
500 IF X=140 THEN 540
510 CALL GCHAR(RR,CR-1,X)
520 IF X=140 THEN 540
530 RETURN
540 REM -HIT
550 HITS=HITS+1
560 CALL HCHAR(RR-1,CR-1,42,3)
570 CALL VCHAR(RR-2,CR,42,3)
580 FOR V=1 TO 30 STEP 4
590 CALL SOUND(100,-5,V)
600 NEXT V
610 CALL CLEAR
620 GOTO 790
630 REM -INITIAL
640 ROCKET$="081C1C1C1C1C3E7F"
650 PLANE$="404040FFFF404040"
660 RANDOMIZE
670 RP=INT(8*RND)+3
680 CP=2
690 RR=24

```

```

700 CR=29
710 CALL CHAR(140,PLANE$)
720 CALL CHAR(150,ROCKET$)
730 CALL HCHAR(RP,CP,140)
740 CALL VCHAR(RR,CR,150)
750 FOR I=1 TO 4
760 CALL SOUND(300,600,2)
770 NEXT I
780 RETURN
790 REM -SCORE
800 CALL CLEAR
810 PRINT "*** SCORE ***"
820 FOR I=1 TO 4
830 CALL SOUND(100,600,2)
840 CALL SOUND(100,800,2)
850 NEXT I
860 PRINT
870 PRINT "NUMBER OF RAIDS=";RAIDS
880 PRINT
890 PRINT "BOMBERS DESTROYED=";HITS
900 IF RAIDS>=10 THEN 960
910 PRINT
920 PRINT
930 PRINT "HIT ENTER TO PREPARE FOR NEXT RAID"
940 INPUT ENTER$
950 GOTO 260
960 PRINT
970 PRINT
980 PRINT "THE WAR IS OVER"
990 CALL SOUND(1000,200,5)
1000 END

```

## Name That Tone

Do you have a good ear for music? Here is a chance to test yourself.

The computer will play seven notes (A, B, C, D, E, F, and G). Repeat these notes as many times as you need in order to memorize them. To repeat the notes, press the "R" key and ENTER.

After you think you have the notes memorized, hit ENTER to start the game. The computer will play any one of the seven notes, at random. Enter the name of the note that you think you just heard.

The objective of the game is to identify correctly 10 notes in a row. Every time you miss, the count starts over, so be careful with your guess.

```

100 REM -NAME THAT TONE
110 FOR I=1 TO 7
120 READ F(I),T$(I)
130 NEXT I
140 DATA 880,A,988,B,1047,C
150 DATA 1175,D,1319,E,1397,F,1568,G
160 CALL CLEAR
170 PRINT "**** NAME THAT TONE ****"
180 PRINT ::
190 PRINT "I AM GOING TO PLAY ALL THE":"NOTES FROM A
    TO G"
200 PRINT :::"TRY TO MEMORIZE EACH TONE"
210 PRINT :::"LATER I WILL PLAY THEM BACK ":"TO SEE IF
    YOU CAN REMEMBER THEM"
220 PRINT :::"CAN YOU IDENTIFY 10 IN A":"ROW?"
230 PRINT :::
240 PRINT "TO PROCEED, HIT ENTER"
250 INPUT RESPONSE$
260 CALL CLEAR
270 FOR I=1 TO 7
280 CALL SOUND(1000,F(I),2)
290 PRINT "THIS IS THE NOTE: ";T$(I)
300 PRINT
310 NEXT I
320 PRINT :::"TO REPEAT, HIT THE R KEY AND ENTER"
330 PRINT :::"TO PROCEED, JUST HIT ENTER"
340 PRINT :::
350 INPUT RESPONSE$
360 IF RESPONSE$="R" THEN 260
370 CALL CLEAR
380 RANDOMIZE
390 N=INT(RND*7)+1
400 PRINT :::::
410 PRINT "WHAT NOTE DID I PLAY"
420 CALL SOUND(700,F(N),2)
430 PRINT "ENTER A,B,C,D,E,F,OR G"
440 INPUT ANS$
450 IF ANS$<"A" THEN 430
460 IF ANS$>"G" THEN 430
470 IF ANS$=T$(N)THEN 540
480 REM -WRONG
490 PRINT
500 PRINT "SORRY - WRONG ANSWER"
510 PRINT "THAT NOTE WAS: ";T$(N)
520 R=0
530 GOTO 580
540 REM -RIGHT
550 PRINT
560 PRINT "CORRECT - VERY GOOD"
570 R=R+1
580 REM -SCORE
590 PRINT
600 PRINT
610 PRINT "CORRECT ANSWERS IN A ROW=";R
620 IF R=10 THEN 630 ELSE 380
630 REM -FINAL SCORE
640 PRINT :::::
650 PRINT "CONGRATULATIONS":::"YOU HAVE WON THE GAME"
660 END

```

## Trampoline

Trampoline is a very interesting program in two ways. First, it is a simple guessing game that pre-school children will find fun and educational, especially those that are just learning how to count. Secondly, it is very interesting from the programming point of view, for it uses special characters to develop a cartoon-like action that shows a man jumping up and down on a trampoline.

*Programming Note* — The special characters are created by the CALL CHAR statements given in lines 190–210. The CALL CHAR statement in line 220 creates the trampoline. The images of the man in various jumping positions are contained in the array called "X" (defined by the DIM statement of line 110).

```
100 REM TRAMPOLINE
110 DIM X(20)
120 X(14)=142
130 X(15)=142
140 X(16)=141
150 X(17)=141
160 X(18)=140
170 X(19)=140
180 C=16
190 CALL CHAR(140,"3838107C92383838")
200 CALL CHAR(141,"383810FE10384482")
210 CALL CHAR(142,"38BA5438927C0000")
220 CALL CHAR(143,"FFFF")
230 CALL CLEAR
240 PRINT "***** TRAMPOLINE *****"
250 PRINT :::
260 GOTO 280
270 CALL CLEAR
280 PRINT "HOW MANY TIMES WILL THE MAN JUMP UP AND
DOWN": "ENTER YOUR GUESS (1 TO 9)"
290 INPUT G
300 IF G>9 THEN 270
310 IF G<1 THEN 270
320 CALL CLEAR
330 CALL HCHAR(20,10,143,14)
340 JUMPS=0
350 RANDOMIZE
360 B=INT(RND*9)+1
370 REM -UP
380 FOR R=19 TO 14 STEP -1
390 CALL VCHAR(R,C,X(R))
400 GOSUB 580
410 NEXT R
420 REM -DOWN
430 FOR R=14 TO 19
440 CALL VCHAR(R,C,X(R))
450 GOSUB 580
```

```

460 NEXT R
470 CALL SOUND(200,-2,2)
480 JUMPS=JUMPS+1
490 J$=STR$(JUMPS)
500 CALL HCHAR(4,25,ASC(J$))
510 IF JUMPS<8 THEN 370
520 CALL CLEAR
530 PRINT "YOU GUESSED ";G
540 PRINT : "HE JUMPED ";JUMPS
550 PRINT :: "TO END, TYPE END" :: "TO REPEAT, HIT ENTER"
560 INPUT RESPONSE$
570 IF RESPONSE$="END" THEN 640 ELSE 270
580 REM DELAY
590 D=10
600 FOR DELAY=1 TO D
610 NEXT DELAY
620 CALL VCHAR(R,C,32)
630 RETURN
640 END

```



## PART III

### **Having Fun With Puzzles and Games**





## Horse Race

In this game, you start with \$500. You then bet all or part of it on one of four horses: A, B, C, or D. If your horse wins, the amount of your bet is added to the \$500. If your horse loses, the amount of your bet is subtracted from \$500. Any time that you have money left, you can bet all or part of it on the next race. The object is to get to \$1000 before you lose all your money.

The horses move randomly so that each race is different. Have fun and good luck.

```
100 REM -HORSE RACE
110 CALL CLEAR
120 PRINT "*** HORSE RACE ***"
130 PRINT
140 PRINT
150 PRINT "IT'S NOW POST TIME"
160 PRINT
170 PRINT "THERE ARE 4 HORSES A, B, C,"
180 PRINT "AND D. ONLY ONE CAN WIN."
190 PRINT
200 PRINT "YOU WILL START WITH $500."
210 PRINT "YOU WILL PICK A, B, C, OR D"
220 PRINT "IF YOU PICK THE WINNER YOUR BET WILL BE
    ADDED."
230 PRINT "IF YOU PICK A LOSER YOUR BET WILL BE
    SUBTRACTED."
240 PRINT
250 PRINT "THE OBJECT IS TO TRY TO GET TO $1000 BEFORE
    GOING BROKE ."
260 PRINT
270 PRINT "GOOD LUCK ."
280 PRINT
290 PRINT "HIT ENTER TO START"
300 INPUT E$
310 Z$(1)="A"
320 Z$(2)="B"
330 Z$(3)="C"
340 Z$(4)="D"
350 R(1)=8
360 R(2)=11
370 R(3)=14
380 R(4)=17
390 HORSE=1
400 AVAIL=500
410 REM =MAIN
420 GOSUB 480
430 GOSUB 540
440 GOSUB 1020
450 GOSUB 760
460 GOSUB 830
470 GOTO 450
480 REM =INITIALIZE
```

```

490 ACOL=5
500 BCOL=5
510 CCOL=5
520 DCOL=5
530 RETURN
540 REM =PLACE BET
550 CALL CLEAR
560 PRINT "AVAILABLE TO BET= $ ";AVAIL
570 PRINT
580 INPUT "ENTER YOUR BET      ":BET
590 IF BET<=AVAIL THEN 630
600 PRINT "YOU BET MORE THAN AVAILABLE"
610 PRINT
620 GOTO 560
630 PRINT
640 INPUT "WHICH HORSE: A, B, C, OR D      ":H$
650 IF ASC(H$)<65 THEN 630
660 IF ASC(H$)>68 THEN 630
670 CALL CLEAR
680 PRINT "THEY'RE OFF"
690 FOR I=1 TO 10
700 PRINT
710 NEXT I
720 FOR I=1 TO 300
730 NEXT I
740 CALL CLEAR
750 RETURN
760 REM =DISPLAY POSITIONS
770 CALL HCHAR(R(HORSE),1,ASC(" "),25)
780 CALL HCHAR(8,ACOL,ASC("A"))
790 CALL HCHAR(11,BCOL,ASC("B"))
800 CALL HCHAR(14,CCOL,ASC("C"))
810 CALL HCHAR(17,DCOL,ASC("D"))
820 RETURN
830 REM =MOVE HORSES
840 RANDOMIZE
850 HORSE=INT(RND*4)+1
860 CALL SOUND(100,800,2)
870 ON HORSE GOTO 880,920,950,980
880 ACOL=ACOL+1
890 IF ACOL<26 THEN 1010
900 IF H$="A" THEN 1080 ELSE 1150
910 REM
920 BCOL=BCOL+1
930 IF BCOL<26 THEN 1010
940 IF H$="B" THEN 1080 ELSE 1150
950 CCOL=CCOL+1
960 IF CCOL<26 THEN 1010
970 IF H$="C" THEN 1080 ELSE 1150
980 DCOL=DCOL+1
990 IF DCOL<26 THEN 1010
1000 IF H$="D" THEN 1080 ELSE 1150
1010 RETURN
1020 REM -TRACK
1030 PRINT "YOU BET $";BET;" ON ";H$
1040 CALL HCHAR(3,1,35,32)
1050 CALL HCHAR(22,1,35,32)
1060 CALL VCHAR(5,26,ASC("."),15)
1070 RETURN
1080 REM =WON RACE

```

```

1090 CALL SOUND(800,150,2)
1100 GOSUB 1320
1110 PRINT
1120 AVAIL=AVAIL+BET
1130 IF AVAIL>=1000 THEN 1220
1140 GOTO 410
1150 REM =LOST RACE
1160 CALL SOUND(800,150,2)
1170 GOSUB 1320
1180 PRINT
1190 AVAIL=AVAIL-BET
1200 IF AVAIL<=0 THEN 1270
1210 GOTO 410
1220 REM =WON THE DAY
1230 CALL CLEAR
1240 PRINT "YOU HAVE WON $ ";AVAIL
1250 PRINT "NICE GOING"
1260 GOTO 1410
1270 REM =WIPED OUT
1280 CALL CLEAR
1290 PRINT "YOU ARE OUT OF MONEY."
1300 PRINT "BETTER LUCK NEXT TIME."
1310 GOTO 1410
1320 REM =PRINT WINNER
1330 CALL CLEAR
1340 PRINT "THE WINNER IS ";Z$(HORSE)
1350 FOR I=1 TO 10
1360 PRINT
1370 NEXT I
1380 FOR I=1 TO 300
1390 NEXT I
1400 RETURN
1410 END

```

## Bomb

The object of this game is to find a bomb before it blows up. The bomb has been placed in a room, at a hotel which has 100 rooms.

A bomb detector will help you find the one room with the bomb. As you check each room, the detector will give you a reading. The closer you are to the room with the bomb, the higher the reading. When you move away from the room with the bomb, the reading will become lower.

You have just 90 seconds to find the bomb. It "costs" you 10 seconds to check a room. Since you only have time to check nine rooms, you must make clever use of your detector.

*Hint:* If you constantly have difficulty with the 90-second time limit, you can increase the time limit by changing line number 110. Change  $T = 90$  to  $T = 120$ , or some other time limit that you may prefer.

```

100 REM -PROGRAM "BOMB"
110 T=90
120 CALL CLEAR
130 PRINT "EMERGENCY !!!"
140 PRINT
150 PRINT "A BOMB HAS BEEN"
160 PRINT "PLACED IN A HOTEL"
170 PRINT "WITH 100 ROOMS."
180 PRINT "YOU HAVE A BOMB"
190 PRINT "DETECTOR TO HELP"
200 PRINT "YOU FIND IT."
210 PRINT
220 PRINT "THE CLOSER YOU GET"
230 PRINT "TO THE BOMB"
240 PRINT "THE HIGHER THE"
250 PRINT "STRENGTH READING."
260 PRINT
270 PRINT "YOU HAVE";T;"SECONDS"
280 PRINT "TO FIND IT"
290 PRINT "BEFORE IT BLOWS UP."
300 PRINT
310 PRINT "GOOD LUCK!"
320 PRINT
330 PRINT "HIT ENTER TO START"
340 INPUT ENTER$
350 CALL CLEAR
360 FOR I=1 TO 3
370 PRINT
380 NEXT I
390 RANDOMIZE
400 B=INT(RND*100)+1
410 PRINT "WHICH ROOM TO SEARCH? (ENTER ROOM NUMBER"
    1-100)"
420 INPUT M
430 PRINT
440 IF M<101 THEN 480
450 PRINT "ONLY 100 ROOMS"
460 PRINT
470 GOTO 410
480 IF M=B THEN 630
490 D=ABS(B-M)
500 V=100-D
510 S=V^2
520 T=T-10
530 IF T=0 THEN 740
540 PRINT "DETECTOR READING= ";S
550 PRINT "SECONDS LEFT= ";T
560 FOR I=1 TO 27
570 PRINT "-";
580 NEXT I
590 PRINT
600 CALL SOUND(500,500,2)
610 GOTO 410
620 GOTO 820
630 REM =FOUND IT
640 CALL CLEAR
650 FOR I=1 TO 7
660 CALL SOUND(-20,150*I,2)
670 NEXT I

```

```

680 PRINT
690 PRINT
700 PRINT "YOU FOUND THE BOMB"
710 PRINT "AND SAVED THE HOTEL"
720 PRINT "YOU ARE A BRAVE HERO"
730 GOTO 820
740 REM =BOOM
750 CALL CLEAR
760 CALL SOUND(1500,120,1)
770 PRINT
780 PRINT
790 PRINT "B O O M !!!"
800 PRINT "THE BOMB BLEW UP"
810 PRINT "IN ROOM ";B
820 END

```

## Remove the X's

Here is a variation of the old matchsticks game, except that you play against the computer.

We will start with 17 X's. You are to remove 1, 2, or 3 X's. The computer will remove 1, 2, or 3. This will proceed until you or the computer is forced to remove the last X. The one that removes the last X loses the game.

```

100 REM -REMOVE X'S
110 CALL CLEAR
120 PRINT "***** REMOVE X'S *****"
130 GOSUB 350
140 PRINT "WE WILL START WITH 17 X'S."
150 PRINT
160 PRINT "YOU WILL REMOVE 1, 2, OR 3 "
170 PRINT "X'S. THE COMPUTER WILL THEN DO THE SAME."
180 PRINT
190 PRINT "THE ONE TO REMOVE THE LAST X LOSES THE
    GAME."
200 PRINT
210 PRINT "CAN YOU BEAT THE COMPUTER?"
220 GOSUB 350
230 PRINT "TO FIND OUT, HIT ENTER TO START."
240 INPUT ENTER$
250 CALL CLEAR
260 RANDOMIZE
270 Q=INT(RND*2)+1
280 LEFT=17
290 GOSUB 400
300 GOSUB 460
310 IF LEFT=0 THEN 790
320 GOSUB 590
330 IF LEFT=0 THEN 840
340 GOTO 290
350 REM -SKIP 3 LINES
360 FOR I=1 TO 3

```

```

370 PRINT
380 NEXT I
390 RETURN
400 CALL CLEAR
410 FOR I=1 TO LEFT
420 PRINT "X";
430 NEXT I
440 PRINT " (<;LEFT;)"
450 RETURN
460 REM -PERSON INPUT
470 GOSUB 350
480 PRINT "HOW MANY DO YOU WANT TO REMOVE? (ENTER 1, 2,
    OR 3)"
490 INPUT P
500 IF P=0 THEN 520
510 IF P<4 THEN 540
520 PRINT "MUST REMOVE 1, 2, OR 3"
530 GOTO 470
540 IF LEFT-P>=0 THEN 570
550 PRINT "CANNOT REMOVE MORE THAN";LEFT
560 GOTO 470
570 LEFT=LEFT-P
580 RETURN
590 REM -COMPUTER MOVE
600 IF Q=1 THEN 680
610 IF LEFT>4 THEN 660
620 C=LEFT-1
630 IF C>0 THEN 690
640 C=1
650 GOTO 690
660 C=INT(RND*3)+1
670 GOTO 690
680 C=4-P
690 PRINT
700 PRINT "YOU HAVE REMOVED ";P
710 FOR I=1 TO 500
720 NEXT I
730 PRINT
740 PRINT "COMPUTER WILL REMOVE";C
750 LEFT=LEFT-C
760 FOR I=1 TO 1000
770 NEXT I
780 RETURN
790 REM -PERSON LOSES
800 CALL CLEAR
810 PRINT "SORRY-YOU LOSE"
820 GOSUB 350
830 GOTO 900
840 REM -COMPUTER LOSES
850 CALL CLEAR
860 PRINT "I LOSE-DARN IT"
870 PRINT
880 PRINT "YOU HAVE BEATEN A COMPUTER"
890 GOSUB 350
900 END

```

## Repeat

Use this game to develop and test your memory.

A one-digit number will flash on the screen. Try to memorize it and enter it on the keyboard. If you are correct, a second digit will be added to the previous digit, and the new two-digit number will be displayed for you to enter as before. This process will continue until you are able to correctly repeat a 10-digit number. If you are able to do this, you will win the game.

Any time you miss entering the correct number, it will be repeated for you. Miss it again and it will be repeated once more. But miss the number three times in a row and you automatically lose. Good luck.

```
100 REM -PROGRAM "REPEAT"
110 CALL CLEAR
120 PRINT "**** REPEAT ****"
130 PRINT
140 PRINT
150 PRINT "TRY TO REPEAT THE"
160 PRINT "NUMBERS THAT YOU SEE"
170 PRINT "ON THE SCREEN."
180 PRINT
190 PRINT "IF YOU GET UP TO"
200 PRINT "10 NUMBERS-YOU WILL WIN."
210 PRINT
220 PRINT "IF YOU MISS 3 TIMES"
230 PRINT "IN A ROW, YOU WILL"
240 PRINT "LOSE. GOOD LUCK!"
250 PRINT
260 PRINT
270 PRINT "HIT ENTER TO START"
280 INPUT A$
290 F=100
300 CALL CLEAR
310 RANDOMIZE
320 X=INT(RND*10)
330 X$=X$&STR$(X)
340 N=N+1
350 MISS=0
360 IF N>=11 THEN 600
370 PRINT TAB(5);X$
380 FOR J=1 TO 10
390 PRINT
400 NEXT J
410 FOR I=1 TO (F*N)
420 NEXT I
430 CALL CLEAR
440 INPUT Q$
450 CALL CLEAR
460 IF Q$=X$ THEN 320
470 MISS=MISS+1
480 IF MISS>=3 THEN 530
```



```

490 PRINT
500 PRINT
510 PRINT
520 GOTO 370
530 REM =LOSE
540 CALL CLEAR
550 PRINT "SORRY-YOU LOSE"
560 PRINT "TIME TO IMPROVE YOUR MEMORY"
570 PRINT
580 PRINT
590 GOTO 690
600 REM =WIN
610 CALL CLEAR
620 PRINT "CONGRATULATIONS"
630 PRINT
640 PRINT "YOU WIN"
650 PRINT
660 PRINT
670 PRINT
680 GOTO 690
690 END

```

## Scramble

In *Scramble*, the computer gives you English words, but with all the letters mixed up. Your job is to guess what the correct word is.

The game is scored so that you get 10 points for each correct answer, and you lose 10 points for each wrong answer. If you don't know the correct word, hit ENTER. The computer will move on to the next word after subtracting 10 points from your score.

The words are scrambled randomly. This means that no two games will play exactly alike, even with the same words. After several games, you may wish to change the words contained in the DATA statements. This is easy to do. However, just make sure that the last word is END, so that the program knows when your list is complete.

```

100 REM -PROGRAM "SCRAMBLE"
110 CALL CLEAR
120 PRINT "**** SCRAMBLE ****"
130 PRINT
140 PRINT
150 PRINT "THE OBJECT OF THIS GAME"
160 PRINT "IS TO UNSCRAMBLE EACH"
170 PRINT "OF THE WORDS."
180 PRINT
190 PRINT "YOU GET 10 POINTS FOR"
200 PRINT "EACH CORRECT ANSWER."
210 PRINT "YOU LOSE 10 POINTS FOR"
220 PRINT "EACH WRONG ANSWER."

```

```

230 PRINT
240 PRINT "IF YOU DON'T KNOW THE"
250 PRINT "ANSWER, HIT THE ENTER KEY."
260 PRINT
270 PRINT
280 PRINT "HIT ENTER NOW TO START."
290 INPUT ENTER$
300 CALL CLEAR
310 DIM W1$(30),W2$(30)
320 CALL CLEAR
330 READ A$
340 IF A$="END" THEN 990
350 GOSUB 560
360 PRINT
370 PRINT
380 PRINT
390 PRINT "UNSCRAMBLE THIS WORD"
400 PRINT "ENTER YOUR ANSWER"
410 PRINT
420 PRINT SCRAM$
430 PRINT
440 INPUT ANSWER$
450 PRINT
460 IF ANSWER$=A$ THEN 750
470 PRINT
480 PRINT "SORRY-YOU LOSE 10 POINTS"
490 PRINT "THE ANSWER IS > ";A$
500 P=P-10
510 PRINT
520 PRINT "YOU NOW HAVE ";P;" POINTS"
530 PRINT
540 GOSUB 880
550 GOTO 330
560 REM =SCRAMBLE RTN
570 SCRAM$=" "
580 TEST$=" "
590 K=0
600 L=LEN(A$)
610 FOR I=1 TO L
620 X(I)=0
630 NEXT I
640 REM
650 RANDOMIZE
660 RAND=INT(RND*L)+1
670 IF X(RAND)=1 THEN 650
680 X(RAND)=1
690 K=K+1
700 SCRAM$=SCRAM$&SEG$(A$,RAND,1)
710 IF K<L THEN 650
720 TEST$=TEST$&A$
730 IF SCRAM$=TEST$ THEN 560
740 RETURN
750 REM = RIGHT
760 P=P+10
770 PRINT "CORRECT-YOU GET 10 POINTS"
780 PRINT "YOU NOW HAVE ";P;" POINTS"
790 GOSUB 880
800 GOTO 330
810 REM =FINAL SCORE
820 FOR I=1 TO 5

```

```

830 PRINT
840 NEXT I
850 PRINT "YOUR FINAL SCORE"
860 PRINT "IS ";P;" POINTS"
870 RETURN
880 REM =SEP LINE
890 FOR I=1 TO 26
900 PRINT "-";
910 NEXT I
920 PRINT
930 RETURN
940 DATA FAT,GAME,SORRY,CUTE
950 DATA FUNNY,SONG,CANDLE,GONE
960 DATA SCHOOL,STREET,RADIO
970 DATA SEEM,PENCIL,QUIET
980 DATA END
990 REM =FINAL SCORE
1000 FOR I=1 TO 5
1010 PRINT
1020 NEXT I
1030 PRINT "YOUR FINAL SCORE"
1040 PRINT "IS ";P;" POINTS."
1050 END

```

## Sorting Batting Averages

Who are the best ranking hitters on your team? This program makes it easy to find out. For each player, enter the name, number of times he came to bat, and the number of hits. After the last player to be ranked, enter 0,0,0 so that the program knows the list is complete.

The program will then compute the batting average of each player. After this, it will rank and display all the averages, listing from highest to lowest.

```

100 REM ="SORTED BATNG AVG"
110 CALL CLEAR
120 PRINT "*** BATTING AVERAGES ***"
130 PRINT
140 PRINT
150 PRINT "THIS PROGRAM COMPUTES"
160 PRINT "A BATTING AVERAGE FOR"
170 PRINT "EACH PLAYER AND FOR"
180 PRINT "THE TEAM. THE AVERAGES"
190 PRINT "ARE THEN SORTED."
200 GOSUB 710
210 GOSUB 710
220 PRINT "HIT ENTER TO START"
230 INPUT ENTER$
240 DIM N$(50),R(50)
250 CALL CLEAR
260 GOSUB 710
270 PRINT "ENTER NAME,AT-BAT,HITS"
280 PRINT "(TO END TYPE 0,0,0)"
290 INPUT NAME$,AB,H

```

```

300 IF AB>=H THEN 350
310 PRINT
320 PRINT "ERROR-MORE HITS THAN AT BATS"
330 PRINT
340 GOTO 270
350 TEAMAB=TEAMAB+AB
360 TEAMH=TEAMH+H
370 IF NAME$="0" THEN 430
380 Z=Z+1
390 AVG=INT(H/AB*1000)
400 N$(Z)=NAME$
410 R(Z)=AVG
420 GOTO 260
430 REM =SORT
440 FOR I=1 TO Z
450 FOR J=I+1 TO Z
460 IF R(I)>R(J) THEN 530
470 H$=N$(I)
480 H2=R(I)
490 N$(I)=N$(J)
500 R(I)=R(J)
510 N$(J)=H$
520 R(J)=H2
530 NEXT J
540 NEXT I
550 CALL CLEAR
560 PRINT "NAME", "AVERAGE"
570 PRINT
580 FOR I=1 TO Z
590 PRINT N$(I), R(I)
600 LINES=LINES+1
610 IF LINES<15 THEN 660
620 PRINT
630 PRINT "HIT ENTER TO CONTINUE"
640 INPUT ENTER$
650 LINES=0
660 NEXT I
670 TAVG=INT(TEAMH/TEAMAB*1000)
680 PRINT
690 PRINT "TEAM AVG", TAVG
700 GOTO 760
710 REM =SKIP 3 LINES
720 FOR I=1 TO 3
730 PRINT
740 NEXT I
750 RETURN
760 END

```

## Dice

Use this program to improve your understanding of dice. When you roll a pair of dice, there are more combinations that add up to the number 7 than there are of any other number. The next highest number of combinations add up to 6 and 8, then to 5 and 9, etc.

You can prove this to yourself by rolling a pair of dice and re-

coding the results. It is easier, however, to simulate this process by using this program. Just type RUN, hit ENTER, and tell the program how many times you want to "roll" the dice. The program will calculate and display the number of times that each possible sum occurs.

While the program is simulating the results, the message, "PLEASE WAIT", will appear on the screen. The greater the number of rolls, the longer the waiting time will be, but the better the results will be.

```
100 REM "DICE ANALYSIS"
110 CALL CLEAR
120 PRINT "*** DICE ANALYSIS ***"
130 PRINT
140 PRINT "THIS PROGRAM HELPS YOU"
150 PRINT "UNDERSTAND HOW DICE WORK"
160 PRINT
170 PRINT "WHEN YOU ADD THE NUMBER"
180 PRINT "OF DOTS ON THE TWO DICE"
190 PRINT "THERE ARE MORE COMBINATIONS"
200 PRINT "OF THE NUMBER 7 THAN ANY"
210 PRINT "OTHER NUMBER."
220 PRINT "THE NEXT MOST FREQUENT"
230 PRINT "NUMBERS THAT OCCUR ARE THE 6 AND 8, ETC."
240 PRINT
250 PRINT "THIS PROGRAM COUNTS HOW MANY"
260 PRINT "TIMES EACH SUM OCCURS."
270 PRINT
280 PRINT "JUST TELL THE PROGRAM"
290 PRINT "HOW MANY TIMES YOU WISH"
300 PRINT "TO ROLL THE DICE."
310 PRINT
320 PRINT
330 PRINT "HIT ENTER TO CONTINUE"
340 INPUT ENTER$
350 CALL CLEAR
360 INPUT "NUMBER OF ROLLS  ":N
370 CALL CLEAR
380 PRINT TAB(7);"PLEASE WAIT"
390 FOR L=1 TO 10
400 PRINT
410 NEXT L
420 DIM X(12)
430 FOR K=1 TO N
440 RANDOMIZE
450 D1=INT(RND*6)+1
460 D2=INT(RND*6)+1
470 T=D1+D2
480 X(T)=X(T)+1
490 NEXT K
500 CALL CLEAR
510 PRINT "SUM", "# OF TIMES"
520 PRINT
530 FOR I=2 TO 12
540 PRINT I,X(I)
550 NEXT I
```

## Paper-Rock-Scissors

Now you can play this favorite hand game against the computer. Choose paper, rock, or scissors. The computer will do the same.

Remember:

Paper covers Rock.

Rock smashes Scissors.

Scissors cut Paper.

Enter your choice of paper, rock, or scissors as P, R, S.

The computer will keep score and tell you who's ahead. To end the program, type END.

```
100 REM -"PAPER-ROCK-SCISSORS"
110 CALL CLEAR
120 PRINT "*** PAPER-ROCK-SCISSORS ***"
130 PRINT
140 PRINT
150 PRINT "CHOOSE PAPER, ROCK OR"
160 PRINT "SCISSORS. THE COMPUTER"
170 PRINT "WILL DO THE SAME."
180 PRINT
190 PRINT
200 PRINT "REMEMBER PAPER COVERS ROCK."
210 PRINT "ROCK SMASHES SCISSORS."
220 PRINT "SCISSORS CUT PAPER."
230 PRINT
240 PRINT
250 PRINT "WHEN ENTERING YOUR CHOICE"
260 PRINT "TYPE P, R, OR S"
270 PRINT
280 PRINT "WHEN FINISHED, TYPE  END."
290 PRINT
300 PRINT
310 PRINT "HIT ENTER TO CONTINUE"
320 PRINT
330 INPUT ENTER$
340 CALL CLEAR
350 INPUT "ENTER P,R,S, OR  END: ";H$
360 IF H$="END" THEN 920
370 FOR K=1 TO 4
380 IF H$=SEG$("PRSX",K,1)THEN 400
390 NEXT K
400 IF K>3 THEN 350
410 GOSUB 560
420 GOSUB 640
430 CALL CLEAR
440 PRINT "YOUR CHOICE WAS- ";H$
450 PRINT "MY CHOICE WAS- ";C$
460 PRINT W$(R(N))
470 GOSUB 870
480 PRINT "YOUR SCORE=";HSCORE
490 PRINT "MY SCORE=";CSCORE
500 FOR I=1 TO 26
```

```

510 PRINT "-";
520 NEXT I
530 GOSUB 870
540 GOSUB 870
550 GOTO 350
560 REM =COMPUTER CHOICE
570 RANDOMIZE
580 RAND=INT(RND*3)+1
590 X$(1)="P"
600 X$(2)="R"
610 X$(3)="S"
620 C$=X$(RAND)
630 RETURN
640 REM =RESULTS & SCORE
650 RESTORE
660 FOR J=1 TO 9
670 READ Q$(J)
680 NEXT J
690 FOR J=1 TO 9
700 READ R(J)
710 NEXT J
720 DATA PP,PR,PS,RP,RR,RS,SP,SR,SS
730 DATA 3,1,2,2,3,1,1,2,3
740 W$(1)="YOU WON"
750 W$(2)="I WON"
760 W$(3)="WE TIED"
770 CALL CLEAR
780 M$=H$&C$
790 FOR N=1 TO 9
800 IF M$=Q$(N) THEN 820
810 NEXT N
820 ON R(N) GOTO 830,850,860
830 HSCORE=HSCORE+1
840 GOTO 860
850 CSCORE=CSCORE+1
860 RETURN
870 REM =SKIP 3 LINES
880 FOR I=1 TO 3
890 PRINT
900 NEXT I
910 RETURN
920 REM =FINAL SCORE
930 CALL CLEAR
940 PRINT "*** FINAL SCORE ***"
950 GOSUB 870
960 PRINT "YOUR SCORE IS ";HSCORE
970 PRINT
980 PRINT "MY SCORE IS ";CSCORE
990 GOSUB 870
1000 GOSUB 870
1010 END

```

## Secret Code

Here is an easy way to write a message in a secret code, where one letter stands for another. Just type in your message and it will appear on the screen in code. The code is contained

in the DATA statements of lines 550 and 560 and may be easily changed.

Here's how it works. The first letter in line 550 converts to the first letter in line 560. The second letter in line 550 converts to the second letter in line 560, etc. Therefore, A is coded as D, B is coded as E, and so on.

You can also use this program to *decode* messages. Just put the DATA statement of line 560 in front of the DATA statement of line 550 (at line 545, for example). Next, delete line 560. Type in your coded message and it will be converted back into English.

```
100 REM -"SECRET CODE"
110 CALL CLEAR
120 PRINT "***** SECRET CODE *****"
130 PRINT
140 PRINT
150 PRINT "THIS PROGRAM CREATES A"
160 PRINT "SECRET MESSAGE. TYPE YOUR"
170 PRINT "WORDS IN ENGLISH AND SEE"
180 PRINT "THEM IN CODE."
190 PRINT
200 PRINT
210 PRINT "YOU CAN CHANGE THE CODE"
220 PRINT "BY CHANGING THE DATA"
230 PRINT "STATEMENTS."
240 PRINT
250 PRINT
260 PRINT "TO END TYPE #"
270 PRINT
280 PRINT
290 PRINT "HIT ENTER TO START"
300 INPUT ENTER$
310 CALL CLEAR
320 DIM E$(50),C$(50),H$(50)
330 FOR I=1 TO 26
340 READ E$(I)
350 NEXT I
360 FOR I=1 TO 26
370 READ C$(I)
380 NEXT I
390 PRINT "BEGIN TYPING MESSAGE"
400 PRINT "<TYPE SLOWLY>"
410 PRINT
420 PRINT
430 CALL KEY(0,KEY,STATUS)
440 IF STATUS=-1 THEN 430
450 IF KEY<32 THEN 430
460 L$=CHR$(KEY)
470 IF L$="#" THEN 570
480 FOR N=1 TO 26
490 IF L$=E$(N) THEN 530
500 NEXT N
510 PRINT L$;
```



```

520 GOTO 430
530 PRINT C$(N) ;
540 GOTO 430
550 DATA A,B,C,D,E,F,G,H,I,J,K,L,M,N,O,P,Q,R,S,T,U,V,
    W,X,Y,Z
560 DATA D,E,F,A,B,C,J,K,L,G,H,I,P,Q,R,M,N,O,V,W,X,S,
    T,U,Z,Y
570 PRINT :::
580 END

```

## Micro Hangman

In this game, the computer will give you a mystery word which you must guess. The computer will tell you how many letters are in the word. Attempt to guess the word by typing in the most likely letter. (Starting with one of the vowels is usually a good idea.) If your letter is in the word, the computer will tell you how many times it occurs and will show you its position. The remaining missing letters will show up as dashes. Continue trying new letters until you think you know what the word is. As soon as you think you know, type in the word GUESS. If you enter the correct guess, the computer will tell you so and will go on to the next word. If your guess is wrong, the old word will reappear.

The objective is to get through all the words while using the fewest number of letters possible. The number in parentheses next to the mystery word indicates how many letters you have tried so far.

The mystery words are contained in the DATA statements and they may be easily changed whenever you wish.

```

100 REM -"MICRO HANGMAN"
110 CALL CLEAR
120 PRINT "**** MICRO HANGMAN ****"
130 PRINT
140 PRINT
150 PRINT "THE COMPUTER WILL PICK"
160 PRINT "A WORD. TRY TO GUESS IT "
170 PRINT "BY ENTERING ONE LETTER AT"
180 PRINT "A TIME. THE COMPUTER WILL"
190 PRINT "TELL YOU IF YOUR LETTER"
200 PRINT "IS IN THE MYSTERY WORD."
210 PRINT
220 PRINT "IF YOU THINK YOU KNOW WHAT"
230 PRINT "THE WORD IS, TYPE THE WORD:  GUESS."
240 PRINT
250 PRINT "IF YOUR GUESS IS CORRECT,"
260 PRINT "THE COMPUTER WILL PICK"
270 PRINT "ANOTHER WORD."

```

```

280 PRINT
290 PRINT
300 PRINT "HIT ENTER TO CONTINUE."
310 INPUT ENTER$
320 CALL CLEAR
330 PRINT
340 PRINT "THE OBJECTIVE IS TO TRY"
350 PRINT "AS FEW LETTERS AS POSSIBLE"
360 PRINT "IN ORDER TO GET THROUGH"
370 PRINT "ALL THE WORDS."
380 PRINT
390 PRINT "AFTER THE LAST WORD YOUR"
400 PRINT "FINAL SCORE WILL BE GIVEN."
410 PRINT "THE BEST SCORE IS THE ONE"
420 PRINT "WITH SMALLEST NUMBER OF TRIES."
430 PRINT
440 PRINT "THE NUMBER IN THE PARENTHESES WILL TELL YOU
    HOW MANY"
450 PRINT "TRIES YOU HAVE HAD SO FAR."
460 PRINT
470 PRINT "HIT ENTER TO START."
480 INPUT ENTER$
490 DIM D$(20)
500 CALL CLEAR
510 GOSUB 1040
520 READ X$
530 IF X$="END" THEN 1090
540 PRINT "I HAVE A MYSTERY WORD"
550 L=LEN(X$)
560 PRINT "IT HAS ";L;" LETTERS"
570 PRINT
580 INPUT "TRY A LETTER OR TYPE:  GUESS ":Q$
590 IF Q$="GUESS" THEN 740
600 NL=NL+1
610 FOR I=1 TO L
620 IF Q$(<)SEG$(X$,I,1)THEN 650
630 D$(I)=Q$
640 M=M+1
650 NEXT I
660 PRINT
670 PRINT "NUMBER FOUND= ";M
680 GOSUB 910
690 M=0
700 PRINT
710 PRINT
720 PRINT
730 GOTO 570
740 REM =GUESS
750 PRINT
760 INPUT "WHAT'S YOUR GUESS: ":ANS$
770 IF ANS$=X$ THEN 830
780 PRINT
790 PRINT "SORRY, THAT'S NOT IT. TRY ANOTHER LETTER."
800 PRINT
810 PRINT
820 GOTO 580
830 REM =CORRECT GUESS
840 PRINT ::"CORRECT-VERY GOOD"
850 M=0
860 PRINT

```

```

870 PRINT
880 PRINT "HIT ENTER FOR NEXT WORD"
890 INPUT ENTER$
900 GOTO 500
910 REM =DISPLAY
920 PRINT
930 PRINT
940 PRINT
950 SW1=1
960 FOR I=1 TO L
970 PRINT D$(I);
980 IF D$(I)="_" THEN 990 ELSE 1000
990 SW1=0
1000 NEXT I
1010 IF SW1=1 THEN 830
1020 PRINT "  (" ;NL;" )"
1030 RETURN
1040 REM =LOAD D$
1050 FOR I=1 TO 20
1060 D$(I)="_"
1070 NEXT I
1080 RETURN
1090 REM =FINAL SCORE
1100 CALL CLEAR
1110 PRINT "  *** FINAL SCORE ***"
1120 PRINT
1130 PRINT
1140 PRINT "TOTAL NUMBER OF "
1150 PRINT "LETTERS TRIED = " ;NL
1160 FOR I=1 TO 10
1170 PRINT
1180 NEXT I
1190 DATA ARM,DOG,HOUSE,SMILE,TEACHER
1200 DATA END
1210 END

```

## Zodiac Signs

Here is a fun way to use your home computer.

Ask a friend to enter his or her birth month and birth date. The computer will then display the right zodiac (astrological sign) for that date. Try it at your next party.

```

100 REM ZODIAC SIGN
110 CALL CLEAR
120 PRINT "*** FIND ZODIAC SIGN ***"
130 PRINT
140 PRINT
150 PRINT "ENTER BIRTHDAY MONTH (1-12)"
160 INPUT M
170 IF M<1 THEN 140
180 IF M>12 THEN 140
190 PRINT
200 PRINT "ENTER BIRTHDAY DATE (1-31)"
210 INPUT D
220 IF D<1 THEN 190

```

```

230 IF D>31 THEN 190
240 N=(M*100)+D
250 RESTORE
260 READ SIGN$,DATE
270 IF SIGN$="END" THEN 570
280 IF N>DATE THEN 260
290 PRINT
300 PRINT
310 PRINT "*****"
320 PRINT
330 PRINT "YOUR ZODIAC SIGN IS :";SIGN$
340 PRINT
350 PRINT "*****"
360 PRINT
370 PRINT
380 PRINT "TO END TYPE  END"
390 PRINT "TO CONTINUE HIT ENTER"
400 INPUT RESPONSE$
410 IF RESPONSE$="END" THEN 570
420 GOTO 130
430 DATA CAPRICORN,0119
440 DATA AQUARIUS,0218
450 DATA PISCES,0320
460 DATA ARIES,0419
470 DATA TAURUS,0520
480 DATA GEMINI,0621
490 DATA CANCER,0722
500 DATA LEO,0822
510 DATA VIRGO,0922
520 DATA LIBRA,1023
530 DATA SCORPIO,1121
540 DATA SAGITTARIO,1221
550 DATA CAPRICORN,1231
560 DATA END,9999
570 END

```



# PART IV

## **Using Educational Programs for Easier Learning**



## Spelling Test

This program can help you improve your spelling. The computer will give you a word. It will spell the word the way it sounds but not necessarily the way it is really spelled. You must then type in the correct spelling. For example:

COMPUTER: Spell REECEEV  
YOU ENTER: RECEIVE

If you spell the word incorrectly, the computer will give you the correct spelling. After the last word, the computer will tell you how many words you got right.

The spelling words are contained in the DATA statements, and may easily be changed.

```
100 REM -"SPELLING TEST"
110 CALL CLEAR
120 PRINT "*** SPELLING TEST ***"
130 GOSUB 360
140 READ PH$,SP$
150 IF PH$="END" THEN 470
160 N=N+1
170 GOSUB 360
180 PRINT "SPELL >";PH$
190 INPUT ANS$
200 IF ANS$=SP$ THEN 290
210 REM =WRONG
220 PRINT
230 PRINT "SORRY WRONG ANSWER"
240 PRINT "THE CORRECT ANSWER IS"
250 PRINT
260 PRINT " >>> ";SP$
270 GOSUB 410
280 GOTO 140
290 REM =RIGHT
300 PRINT
310 PRINT
320 PRINT "CORRECT-VERY GOOD"
330 RIGHT=RIGHT+1
340 GOSUB 410
350 GOTO 140
360 REM =SKIP 3 LINES
370 FOR I=1 TO 4
380 PRINT
390 NEXT I
400 RETURN
410 REM =DASHES
420 FOR I=1 TO 26
430 PRINT "-";
440 NEXT I
450 PRINT
460 RETURN
470 REM = FINAL SCORE
```



```

480 PCNT=INT(RIGHT/N*1000)/10
490 CALL CLEAR
500 PRINT "----- FINAL SCORE ----"
510 GOSUB 360
520 GOSUB 360
530 PRINT "YOU SPELLED CORRECTLY"
540 PRINT RIGHT;"WORDS OUT OF ";N
550 PRINT
560 PRINT "THAT'S ";PCNT;"%"
570 GOSUB 360
580 END
590 DATA ADRES,ADDRESS
600 DATA REECEEV,RECEIVE
610 DATA HITE,HEIGHT
620 DATA BIZEE,BUSY
630 DATA FRAYT,FREIGHT
640 DATA NAYBOR,NEIGHBOR
650 DATA LAFF,LAUGH
660 DATA BYURDH,BUREAU
670 DATA KOMITEE,COMMITTEE
680 DATA END,END

```

## Coin Count

Here is a good way to teach a child the value of money or, more specifically, the value of coins.

The computer will tell the child how many coins it has in its "pocket" (the number of halves, quarters, dimes, nickels, and pennies). The child is asked to figure out the value of all these coins. If his answer is correct, this amount is added to the child's total. If the answer is wrong, the correct amount is subtracted from the child's total. The total, however, does not drop below zero.

The objective is for the child to reach a total of \$10.00.

```

100 REM -"COIN COUNT"
110 GOAL=10
120 GOSUB 520
130 READ NH,NQ,ND,NN,NP
140 RANDOMIZE
150 RH=INT(RND*NH)+1
160 RQ=INT(RND*NQ)+1
170 RD=INT(RND*ND)+1
180 RN=INT(RND*NN)+1
190 RP=INT(RND*NP)+1
200 T=(RH*50)+(RQ*25)+(RD*10)+(RN*5)+(RP*1)
210 T=T/100
220 GOSUB 730
230 IF ANS=T THEN 360
240 REM =WRONG
250 PRINT
260 PRINT
270 PRINT "SORRY-THAT'S WRONG"

```

```

280 PRINT "THE CORRECT ANSWER IS $";T
290 T9=T9-T
300 IF T9>0 THEN 390
310 T9=0
320 PRINT "YOU NOW HAVE $";T9
330 FOR I=1 TO 1000
340 NEXT I
350 GOTO 220
360 REM =RIGHT
370 PRINT
380 T9=T9+T
390 PRINT "CORRECT-YOU EARN $";T
400 PRINT
410 PRINT "YOU NOW HAVE $";T9
420 FOR I=1 TO 1000
430 NEXT I
440 IF T9<10 THEN 140
450 REM =FINAL
460 FOR K=1 TO 500
470 NEXT K
480 CALL CLEAR
490 PRINT "CONGRATULATIONS"
500 PRINT "YOU HAVE REACHED THE GOAL OF $";GOAL
510 GOTO 910
520 REM =RULES
530 CALL CLEAR
540 PRINT TAB(7);"COIN COUNT"
550 PRINT
560 PRINT
570 PRINT "IN MY POCKET I HAVE SOME"
580 PRINT "COINS"
590 PRINT
600 PRINT "IF YOU FIGURE OUT HOW MUCH THEY ARE WORTH
    IN DOLLARS, THIS AMOUNT WILL BE ADDED TO YOUR TOTAL"
610 PRINT
620 PRINT "IF YOU GUESS WRONG THIS"
630 PRINT "AMOUNT WILL BE TAKEN AWAY"
640 PRINT "FROM YOUR TOTAL"
650 PRINT
660 PRINT "AS SOON AS YOU REACH $";GOAL
670 PRINT "YOU WILL WIN"
680 PRINT
690 PRINT
700 PRINT "HIT ENTER TO CONTINUE"
710 INPUT ENTER$
720 RETURN
730 REM -SCREEN
740 CALL CLEAR
750 PRINT "IN MY POCKET I HAVE:"
760 PRINT
770 PRINT
780 PRINT
790 PRINT RH;"--HALVES"
800 PRINT RQ;"--QUARTERS"
810 PRINT RD;"--DIMS"
820 PRINT RN;"--NICKELS"
830 PRINT RP;"--PENNIES"
840 PRINT
850 PRINT
860 PRINT "HOW MUCH MONEY DO I HAVE?"

```

```

870 PRINT
880 INPUT "ENTER IN DOLLARS $":ANS
890 RETURN
900 DATA 3,3,5,5,9
910 END

```

## Checking Math Homework

Students in the early grades (first through sixth) can use this program to check and grade their homework.

Unlike a calculator, this program will not give the answer to the problem. Instead, it will only tell the student if his answer is correct or incorrect. If his answer is incorrect, the student will have to rework the problem since he does not know what the answer is.

Addition, subtraction, multiplication, and division problems are the types that can be checked. Scoring is automatic and a final score is given at the end.

```

100 REM -"CHECK MATH HOMEWORK"
110 CALL CLEAR
120 PRINT "    CHECK MATH HOMEWORK"
130 PRINT
140 PRINT
150 PRINT "THIS PROGRAM CHECKS AND"
160 PRINT "GRADES YOUR MATH HOMEWORK"
170 PRINT
180 PRINT "AFTER THE ? TYPE IN"
190 PRINT "THE FIRST NUMBER(A),"
200 PRINT "THE SECOND NUMBER (B)"
210 PRINT "AND THE ANSWER"
220 PRINT
230 PRINT "WHEN YOU FINISH"
240 PRINT "TYPE 0,0,0 AFTER THE ?"
250 GOSUB 740
260 PRINT "TO CONTINUE HIT ENTER"
270 INPUT ENTER$
280 CALL CLEAR
290 PRINT "WHICH MATH OPERATION DO YOU WANT TO CHECK?"
300 GOSUB 740
310 PRINT "ADD (A+B) --- 1"
320 PRINT "SUB (A-B) --- 2"
330 PRINT "MUL (A*B) --- 3"
340 PRINT "DIV (A/B) --- 4"
350 GOSUB 740
360 INPUT "ENTER 1, 2, 3, OR 4 ":OPER
370 CALL CLEAR
380 REM =MAIN RTN
390 GOSUB 740
400 PRINT "ENTER 1ST NUMBER, 2ND NUMBER, ANSWER"
410 INPUT A,B,C
420 IF A+B+C=0 THEN 580

```

```

430 ON OPER GOTO 660,680,700,720
440 N=N+1
450 IF C=ANS THEN 520
460 PRINT
470 PRINT "SORRY - WRONG ANSWER"
480 PRINT "YOUR SCORE IS NOW"
490 PRINT SCORE;"OUT OF";N
500 PRINT
510 GOTO 380
520 REM =CORRECT ANS
530 SCORE=SCORE+1
540 PRINT
550 PRINT "CORRECT-YOUR SCORE IS NOW"
560 PRINT SCORE;"OUT OF ";N
570 GOTO 380
580 REM =FINAL SCORE
590 PCNT=INT(SCORE/N*1000)/10
600 CALL CLEAR
610 PRINT "YOUR FINAL SCORE IS ";SCORE;" OUT OF ";N
620 PRINT
630 PRINT "WHICH IS ";PCNT;" %"
640 GOTO 790
650 REM =OPERATIONS
660 ANS=A+B
670 GOTO 440
680 ANS=A-B
690 GOTO 440
700 ANS=A*B
710 GOTO 440
720 ANS=A/B
730 GOTO 440
740 REM =3 LINES
750 FOR I=1 TO 3
760 PRINT
770 NEXT I
780 RETURN
790 END

```

## Guess the Missing Letters

In this word recognition game, the computer will give you a word that has two of the letters missing. For example:

M-L-IPLY

Your task is to figure out the missing letters which, in this case, are "U" and "T". Type in "U" and "T" (remember to insert the comma between the letters).

If you guess correctly, you earn 10 points. If your guess is wrong, you get two more chances. If you still can't get the answer after three tries, the computer will give you another word. The objective is to get as high a score as possible. At the end of the game, the computer will tell you your final score.

Words and their missing letters are stored in the DATA statements, and may easily be changed. The last DATA statement must be END,X,X.

```
100 REM -"MISSING LETTERS"
110 CALL CLEAR
120 PRINT "GUESS THE MISSING LETTERS"
130 PRINT
140 PRINT
150 PRINT
160 PRINT
170 PRINT "FOR EACH CORRECT ANSWER"
180 PRINT "YOU GET 10 POINTS"
190 PRINT
200 PRINT "AFTER 3 MISSES A NEW WORD"
210 PRINT "WILL APPEAR"
220 PRINT
230 PRINT "TRY TO GET AS HIGH A SCORE"
240 PRINT "AS POSSIBLE"
250 PRINT
260 PRINT "PLEASE ENTER LETTERS"
270 PRINT "SEPARATED BY COMMAS (X,X)"
280 PRINT
290 PRINT
300 PRINT "HIT ENTER TO CONTINUE"
310 INPUT ENTER$
320 CALL CLEAR
330 RESTORE
340 GOSUB 520
350 READ A$,B$,C$
360 ERROR=0
370 IF A$="END" THEN 690
380 PRINT TAB(9);A$
390 PRINT
400 PRINT
410 PRINT "GUESS THE MISSING LETTERS"
420 GOSUB 560
430 PRINT "ENTER LETTERS (X,X)"
440 INPUT X$,Y$
450 IF X$(<>)B$ THEN 600
460 IF Y$(<>)C$ THEN 600
470 GOSUB 520
480 P=P+10
490 PRINT "CORRECT-YOU HAVE";P;"POINTS"
500 GOSUB 520
510 GOTO 350
520 FOR I=1 TO 5
530 PRINT
540 NEXT I
550 RETURN
560 FOR I=1 TO 10
570 PRINT
580 NEXT I
590 RETURN
600 REM =WRONG
610 ERROR=ERROR+1
620 GOSUB 520
```

```

630 PRINT "SORRY-WRONG ANSWER"
640 ON ERROR GOTO 650,650,340
650 PRINT "MISS # ";ERROR
660 PRINT
670 GOSUB 520
680 GOTO 380
690 GOSUB 520
700 PRINT "   ***   FINAL SCORE   ***"
710 PRINT
720 PRINT
730 PRINT "TOTAL POINTS=";P
740 GOSUB 520
750 END
760 DATA M-L-IPLY,U,T
770 DATA FO-E-T,R,S
780 DATA FR-E-D,I,N
790 DATA S-S-ER,I,T
800 DATA S-HO-L,C,O
810 DATA P--TY,A,R
820 DATA KI--HEN,T,C
830 DATA -AD-O,R,I
840 DATA GU-T-R,I,A
850 DATA -I-RARY,L,B
860 DATA T-A-HER,E,C
870 DATA -IA-O,P,N
880 DATA END,X,X

```

## Savings Account

The benefits of regular weekly savings are illustrated by this program. Just answer several questions about your savings plan and the computer will project the value of your account for 10 years into the future.

You may want to test several different savings plans to see which is the best one for you.

```

100 REM -"SAVINGS ACCOUNT"
110 CALL CLEAR
120 PRINT "**** SAVINGS ACCOUNT ****"
130 PRINT
140 PRINT
150 INPUT "HOW MUCH DO YOU WANT TO SAVE EVERY WEEK? $":WS
160 PRINT
170 INPUT "AVG ANNUAL INT RATE %=" ":I
180 PRINT
190 INPUT "HOW MUCH IS IN YOUR ACCOUNT NOW? $":BAL
200 PRINT
210 INPUT "STARTING YEAR ":YEAR
220 CALL CLEAR
230 PRINT "***WATCH YOUR SAVINGS GROW***"
240 PRINT
250 PRINT
260 PRINT " YEAR", "BALANCE"

```

```

270 PRINT
280 YS=WS*52
290 FOR N=1 TO 10
300 BAL=BAL+YS+(1/100*(BAL+YS/2))
310 BAL=INT(BAL)
320 PRINT (YEAR+N),BAL
330 NEXT N
340 END

```

## Planets

This two-part exercise should help expand your knowledge of our solar system.

In Part I, you are asked to name all the planets, in order, from the sun outward. Whenever you answer incorrectly, you will be given another chance to get it right. If you get stuck, enter the word **HELP** and the computer will give you the correct name.

After you have named all the planets, Part II of the program will ask you for your weight. The computer will then tell you how much you would weigh on all the other planets.

```

100 REM -"PLANETS"
110 CALL CLEAR
120 PRINT "***** PLANETS *****"
130 PRINT
140 PRINT
150 PRINT "NAME THE PLANETS IN "
160 PRINT "ORDER FROM THE SUN"
170 GOSUB 470
180 READ PLANET$,FACT
190 IF PLANET$="END" THEN 680
200 GOSUB 520
210 GOSUB 470
220 PRINT "NAME THE NEXT PLANET "
230 INPUT ANS$
240 IF ANS$="HELP" THEN 400
250 IF ANS$(<>)PLANET$ THEN 310
260 PRINT
270 PRINT
280 PRINT "CORRECT-VERY GOOD"
290 GOSUB 470
300 GOTO 180
310 REM =WRONG
320 GOSUB 470
330 PRINT "SORRY-THAT'S WRONG"
340 PRINT "KEEP TRYING"
350 PRINT
360 PRINT "IF YOU STILL CAN'T THINK"
370 PRINT "OF IT, TYPE HELP"
380 PRINT
390 GOTO 200
400 GOSUB 470

```

```

410 PRINT "THE ANSWER IS >>> ";PLANET$
420 PRINT
430 PRINT
440 PRINT
450 GOTO 180
460 RETURN
470 REM =SKIP 3 LINES
480 FOR I=1 TO 3
490 PRINT
500 NEXT I
510 RETURN
520 REM =DASHES
530 FOR I=1 TO 27
540 PRINT "-";
550 NEXT I
560 PRINT
570 RETURN
580 DATA MERCURY,0.28
590 DATA VENUS,0.85
600 DATA EARTH,1
610 DATA MARS,0.38
620 DATA JUPITER,2.6
630 DATA SATURN,1.2
640 DATA URANUS,1.1
650 DATA NEPTUNE,1.4
660 DATA PLUTO,.6
670 DATA END,0
680 REM =WEIGHTS
690 FOR I=1 TO 700
700 NEXT I
710 CALL CLEAR
720 PRINT "PLEASE ENTER YOUR WEIGHT"
730 PRINT "ON EARTH AND FIND OUT"
740 PRINT "HOW MUCH YOU WEIGH ON OTHER PLANETS"
750 GOSUB 470
760 INPUT "ENTER YOUR WEIGHT  ":WEARTH
770 CALL CLEAR
780 PRINT "PLANET","WEIGHT"
790 GOSUB 470
800 RESTORE
810 READ PLANET$,FACT
820 IF PLANET$="END" THEN 860
830 PW=INT(WEARTH*FACT)
840 PRINT PLANET$,PW
850 GOTO 810
860 END

```

## Temperatures

This program will help you gain a better understanding of the Celsius temperature scale, which is used in the metric system. It does this in two ways:

1. It shows you a table of equivalent temperatures in both the Fahrenheit scale and the Celsius scale.



2. It provides an easy way for you to convert from Fahrenheit to Celsius. Just enter any Fahrenheit temperature and the computer will give you the equivalent Celsius temperature.

```
100 REM --"TEMPERATURES"
110 CALL CLEAR
120 PRINT "**** TEMPERATURES ****"
130 GOSUB 330
140 PRINT "THIS PROGRAM WILL HELP YOU"
150 PRINT "BECOME FAMILIAR WITH"
160 PRINT "THE CELSIUS TEMPERATURE"
170 PRINT "SCALE BY PRODUCING"
180 PRINT "A TABLE OF EQUIVALENT"
190 PRINT "TEMPERATURES IN"
200 PRINT "FAHRENHEIT AND CELSIUS."
210 GOSUB 330
220 PRINT "HIT ENTER TO CONTINUE"
230 INPUT ENTER$
240 CALL CLEAR
250 PRINT "FAHRENHEIT", "CELSIUS"
260 PRINT
270 FOR F=-20 TO 120 STEP 10
280 C=(F-32)*(5/9)
290 C=INT(C*10)/10
300 PRINT F,C
310 NEXT F
320 GOTO 380
330 REM =SKIP 3 LINES
340 FOR I=1 TO 3
350 PRINT
360 NEXT I
370 RETURN
380 GOSUB 330
390 PRINT "HIT ENTER TO CONTINUE"
400 INPUT ENTER$
410 CALL CLEAR
420 REM =CONVERT F TO C
430 PRINT "THE NEXT PART OF THIS PROGRAM WILL CONVERT
    SINGLE FAHRENHEIT TEMPERATURES TO CELSIUS
    TEMPERATURES."
440 PRINT
450 PRINT "HIT ENTER TO START"
460 INPUT ENTER$
470 CALL CLEAR
480 PRINT "INPUT FAHRENHEIT DEGREES (TO END TYPE -9999)"
490 INPUT F
500 IF F=-9999 THEN 580
510 C=5*(F-32)/9
520 C=INT(C*10)/10
530 GOSUB 330
540 PRINT F;"DEGREES F=";C;"DEGREES C"
550 GOSUB 330
560 GOSUB 330
570 GOTO 480
580 END
```

## Speed Reading

This program is designed to help you learn how to read better and faster. Sentences will be flashed on the screen. Try to read each sentence at a single glance rather than one word at a time. After some practice, you should be able to increase the speed at which the sentences are presented and should still be able to understand the meaning.

The speed at which the sentences are flashed is controlled by a number from 1 (fastest) to 10 (slowest). At the start, the computer will ask you to enter this number. You may want to experiment with different speeds to see which is best for you.

```
100 REM --"SPEED READING"
110 CALL CLEAR
120 PRINT "**** SPEED READING ****"
130 PRINT
140 PRINT
150 PRINT "THIS PROGRAM WILL"
160 PRINT "HELP YOU READ FASTER"
170 PRINT
180 PRINT "THE TEXT IS IN THE DATA"
190 PRINT "STATEMENTS AND MAY BE EASILY CHANGED"
200 PRINT
210 PRINT
220 PRINT "READING SPEEDS RANGE FROM 1=FASTEST TO
    10=SLOWEST"
230 PRINT
240 PRINT
250 PRINT
260 INPUT "ENTER READING SPEED(1 TO 10)":S
270 IF S<1 THEN 250
280 IF S>10 THEN 250
290 F=-2
300 SEC=(2*S)/10*1000
310 CALL SOUND(SEC,F,30)
320 READ A$
330 IF A$="XXX" THEN 460
340 CALL CLEAR
350 PRINT A$
360 CALL SOUND(SEC,F,30)
370 GOTO 320
380 DATA "SNOWFLAKES ARE TINY CRYSTALS"
390 DATA "OF ICE. THEY FORM WHEN WATER"
400 DATA "VAPOR FREEZES. WHEN THEY HIT"
410 DATA "THE GROUND THEY BLANKET OUR"
420 DATA "ROADS AND HIGHWAYS AND MAKE"
430 DATA "IT DIFFICULT TO DRIVE. BUT"
440 DATA "THEY TRULY ARE BEAUTIFUL."
450 DATA XXX
460 CALL CLEAR
470 END
```

## State Capitals

Here is a fun way to learn and memorize the capitals of our fifty states.

The computer will randomly pick ten state capitals. Your job is to name the state whose capital is shown on the screen. At the end, the computer will tell you how many capitals you named correctly and will give your percentage score.

Since the states are picked randomly, the order of the state capitals should be different each time you run the program.

```
100 REM -"STATE CAPITALS"
110 CALL CLEAR
120 PRINT TAB(7);"STATE CAPITALS"
130 GOSUB 460
140 DIM STATE$(55),CAP$(55)
150 READ S$,C$
160 IF S$="END" THEN 210
170 N=N+1
180 STATE$(N)=S$
190 CAP$(N)=C$
200 GOTO 150
210 REM =NEXT STATE
220 GOSUB 510
230 Z=Z+1
240 IF Z>=11 THEN 1080
250 GOSUB 460
260 PRINT "NAME THE STATE WHOSE CAPITAL IS: "
270 RANDOMIZE
280 R=INT(RND*N)+1
290 IF CAP$(R)=" " THEN 280
300 PRINT CAP$(R)
310 INPUT A$
320 IF A$=STATE$(R) THEN 370
330 PRINT
340 PRINT
350 PRINT "SORRY-THE ANSWER IS ";STATE$(R)
360 GOTO 210
370 REM =CORRECT
380 GOSUB 460
390 PRINT "CORRECT - VERY GOOD"
400 K=K+1
410 PRINT "YOU NOW HAVE:"
420 PRINT K;" CORRECT ANSWERS"
430 STATE$(R)=" "
440 CAP$(R)=" "
450 GOTO 210
460 REM =SKIP 3 LINES
470 FOR I=1 TO 3
480 PRINT
490 NEXT I
500 RETURN
510 REM =DASHES
520 FOR I=1 TO 27
```

```

530 PRINT "-";
540 NEXT I
550 PRINT
560 RETURN
570 DATA ALABAMA,MONTGOMERY
580 DATA ALASKA,JUNEAU
590 DATA ARIZONA,PHOENIX
600 DATA ARKANSAS,LITTLE ROCK
610 DATA CALIFORNIA,SACRAMENTO
620 DATA COLORADO,DENVER
630 DATA CONNECTICUT,HARTFORD
640 DATA DELAWARE,DOVER
650 DATA FLORIDA,TALLAHASSEE
660 DATA GEORGIA,ATLANTA
670 DATA HAWAII,HONOLULU
680 DATA IDAHO,BOISE
690 DATA ILLINOIS,SPRINGFIELD
700 DATA INDIANA,INDIANAPOLIS
710 DATA IOWA,DES MOINES
720 DATA KANSAS,TOPEKA
730 DATA KENTUCKY,FRANKFORT
740 DATA LOUISIANA,BATON ROUGE
750 DATA MAINE,AUGUSTA
760 DATA MARYLAND,ANNAPOLIS
770 DATA MASSACHUSETTS,BOSTON
780 DATA MICHIGAN,LANSING
790 DATA MINNESOTA,ST PAUL
800 DATA MISSISSIPPI,JACKSON
810 DATA MISSOURI,JEFFERSON CITY
820 DATA MONTANA,HELENA
830 DATA NEBRASKA,LINCOLN
840 DATA NEVADA,CARSON CITY
850 DATA NEW HAMPSHIRE,CONCORD
860 DATA NEW JERSEY,TRENTON
870 DATA NEW MEXICO,SANTA FE
880 DATA NEW YORK,ALBANY
890 DATA NORTH CAROLINA,RALEIGH
900 DATA NORTH DAKOTA,BISMARCK
910 DATA OHIO,COLUMBUS
920 DATA OKLAHOMA,OKLAHOMA CITY
930 DATA OREGON,SALEM
940 DATA PENNSYLVANIA,HARRISBURG
950 DATA RHODE ISLAND,PROVIDENCE
960 DATA SOUTH CAROLINA,COLUMBIA
970 DATA SOUTH DAKOTA,PIERRE
980 DATA TENNESSEE,NASHVILLE
990 DATA TEXAS,AUSTIN
1000 DATA UTAH,SALT LAKE CITY
1010 DATA VERMONT,MONTPELIER
1020 DATA VIRGINIA,RICHMOND
1030 DATA WASHINGTON,OLYMPIA
1040 DATA WEST VIRGINIA,CHARLESTON
1050 DATA WISCONSIN,MADISON
1060 DATA WYOMING,CHEYENNE
1070 DATA END,END
1080 REM =TOTAL SCORE
1090 Z=Z-1
1100 GOSUB 460
1110 GOSUB 460
1120 PRINT TAB(7);"FINAL SCORE"

```

```

1130 GOSUB 460
1140 PRINT "YOU GUESSED ";K;" OUT OF ";Z
1150 SCORE=INT(K/Z*100)
1160 PRINT "YOUR SCORE IS ";SCORE;" %"
1170 END

```

## English to Metric

This program can be used to take some of the mystery out of metrics. Just tell the computer which measure you want to covert:

Inches to centimeters  
Pounds to grams  
Gallons to liters  
etc.

Then enter your quantity and the computer will do the rest.

```

100 REM -"ENGLISH TO METRIC"
110 CALL CLEAR
120 PRINT "***** ENGLISH TO METRIC *****"
130 PRINT
140 PRINT
150 PRINT
160 PRINT "INCHES TO CENTIMETERS -- 1"
170 PRINT "METERS TO YARDS ----- 2"
180 PRINT "MILES TO KILOMETERS --- 3"
190 PRINT "OUNCES TO GRAMS ----- 4"
200 PRINT "POUNDS TO GRAMS ----- 5"
210 PRINT "GALLONS TO LITERS----- 6"
220 PRINT
230 PRINT
240 PRINT
250 PRINT "PLEASE ENTER YOUR CHOICE."
260 PRINT
270 PRINT
280 INPUT "ENTER 1 THROUGH 6 ":N
290 IF N>6 THEN 270
300 IF N<1 THEN 270
310 CALL CLEAR
320 ON N GOSUB 360,410,460,510,560,610
330 PRINT
340 PRINT
350 GOTO 120
360 INPUT "ENTER INCHES ":IN
370 GOSUB 660
380 PRINT IN;" INCHES=";2.54*IN;" CENTIMETER"
390 GOSUB 710
400 RETURN
410 INPUT "ENTER METERS ":M
420 GOSUB 660
430 PRINT M;" METERS=";1.0936*M;" YARDS"
440 GOSUB 710

```

```

450 RETURN
460 INPUT "ENTER MILES ":M
470 GOSUB 660
480 PRINT M;"MILES=";1.609*M;"KILOMETERS"
490 GOSUB 710
500 RETURN
510 INPUT "ENTER OUNCES ":Z
520 GOSUB 660
530 PRINT Z;"OUNCES=";28.35*Z;"GRAMS"
540 GOSUB 710
550 RETURN
560 INPUT "ENTER POUNDS ":P
570 GOSUB 660
580 PRINT P;"POUNDS=";453.6*P;"GRAMS"
590 GOSUB 710
600 RETURN
610 INPUT "ENTER GALLONS ":G
620 GOSUB 660
630 PRINT G;"GALLONS=";3.785*G;"LITERS"
640 GOSUB 710
650 RETURN
660 REM =SKIP 3 LINES
670 FOR I=1 TO 3
680 PRINT
690 NEXT I
700 RETURN
710 REM =RETURN RTN
720 FOR I=1 TO 10
730 PRINT
740 NEXT I
750 PRINT "HIT ENTER TO CONTINUE "
760 PRINT "OR TYPE END TO STOP."
770 INPUT E$
780 IF E$="END" THEN 800
790 GOTO 110
800 END

```

## Area of a Rectangle

This program computes the area of a rectangle and then draws a picture of the rectangle on the screen.

Any reasonable height and width can be entered. The computer will automatically scale the picture to fit the screen.

```

100 REM -AREA OF A RECTANGLE
110 CALL CLEAR
120 PRINT "*** AREA OF A RECTANGLE ***"
130 PRINT
140 PRINT
150 PRINT
160 PRINT "THIS PROGRAM COMPUTES THE AREA OF A
    RECTANGLE FOR ANY HEIGHT AND WIDTH."
170 PRINT
180 PRINT "THE RECTANGLE IS THEN DRAWN TO SCALE IN THE
    CENTER OF THE SCREEN."

```

```

190 PRINT
200 PRINT
210 PRINT "HIT ENTER TO START"
220 INPUT ENTER$
230 CALL CHAR(140,"FFFFFFFFFFFFFF")
240 CALL CLEAR
250 PRINT "ENTER RECTANGLE HEIGHT AND    WIDTH (H,W). "
260 INPUT HI,WI
270 GOSUB 510
280 H=INT(HI/SCALE)
290 W=INT(WI/SCALE)
300 IF H>2 THEN 320
310 H=2
320 IF W>2 THEN 340
330 W=2
340 CALL CLEAR
350 R=14-H/2
360 C=16-W/2
370 X=140
380 CALL HCHAR(R,C,X,W)
390 CALL HCHAR(R+H,C,X,W+1)
400 CALL VCHAR(R,C,X,H)
410 CALL VCHAR(R,C+W,X,H)
420 PRINT "AREA=";HI;" X ";WI;"=";HI*WI
430 PRINT "DRAWING SCALE=";SCALE
440 PRINT
450 PRINT "TO CONTINUE, HIT ENTER."
460 PRINT "TO END, HIT THE E KEY."
470 CALL KEY(0,K,S)
480 IF K=69 THEN 570
490 IF K>0 THEN 240
500 GOTO 470
510 REM -SCALE
520 IF HI<WI THEN 550
530 SCALE=INT(HI/15)+1
540 GOTO 560
550 SCALE=INT(WI/15)+1
560 RETURN
570 END

```

## Language Translator

Here is an interesting way to practice and learn the words of a foreign language.

This program makes it possible for you to translate any of the English words contained in its vocabulary into French and Spanish. Just indicate how you wish to translate (English to French or English to Spanish) and give the computer the word you wish to translate. The translated word will appear on the screen next to your word.

The example vocabulary used in the program contains 14 words. This vocabulary may be easily changed or expanded. You may even want to adapt it to other languages. All this can be easily done by just changing the DATA statements.

```

100 REM  LANGUAGE TRANSLATOR
110 CALL CLEAR
120 DIM LANG$(2,200),ENG$(200)
130 READ L$(1),L$(2)
140 GOSUB 190
150 GOSUB 250
160 GOSUB 410
170 GOSUB 620
180 GOTO 160
190 REM -READ VOCABULARY
200 N=N+1
210 READ ENG$(N),LANG$(1,N),LANG$(2,N)
220 IF ENG$(N)="END" THEN 240
230 GOTO 200
240 RETURN
250 REM -TRANSLATION CHOICE
260 CALL CLEAR
270 PRINT "---- LANGUAGE TRANSLATOR ----"
280 GOSUB 820
290 GOSUB 820
300 PRINT "HOW WOULD YOU LIKE TO TRANSLATE?"
310 GOSUB 820
320 PRINT "1-ENGLISH TO ";L$(1)
330 PRINT "2-ENGLISH TO ";L$(2)
340 GOSUB 820
350 GOSUB 820
360 PRINT "PLEASE ENTER CHOICE. TYPE 1 OR 2 AND HIT
    ENTER."
370 INPUT CHOICE
380 IF CHOICE>2 THEN 340
390 IF CHOICE<0 THEN 340
400 RETURN
410 REM -WORD TO TRANSLATE
420 CALL CLEAR
430 L=0
440 PRINT "*** MY VOCABULARY ***"
450 PRINT
460 L=0
470 FOR I=1 TO N-1
480 PRINT ENG$(I)
490 L=L+1
500 IF L<9 THEN 550
510 PRINT
520 PRINT "HIT ENTER TO CONTINUE"
530 INPUT RESPONSE$
540 L=0
550 NEXT I
560 GOSUB 820
570 PRINT "WHICH WORD WOULD YOU LIKE TO TRANSLATE TO ";
    L$(CHOICE)
580 PRINT "(TO CHANGE LANGUAGES, ENTER: CHANGE). "
590 INPUT WORD$
600 IF WORD$="CHANGE" THEN 150
610 RETURN
620 REM -TRANSLATE
630 CALL CLEAR
640 FOR I=1 TO N-1
650 IF WORD$(<)>ENG$(I) THEN 760
660 PRINT "ENGLISH",L$(CHOICE)
670 PRINT "-----", "-----"

```



```

480 PRINT
490 PRINT WORD$,LANG$(CHOICE,I)
700 GOSUB 820
710 PRINT "TO END, TYPE: END,"
720 PRINT "TO CONTINUE, HIT ENTER."
730 INPUT RESPONSE$
740 IF RESPONSE$="END" THEN 1030
750 GOTO 810
760 NEXT I
770 GOSUB 820
780 PRINT WORD$;"- NOT IN MY VOCABULARY."
790 PRINT
800 GOTO 440
810 RETURN
820 REM -SKIP 3 LINES
830 FOR Z=1 TO 3
840 PRINT
850 NEXT Z
860 RETURN
870 DATA FRENCH,SPANISH
880 DATA LOVE,AMOUR,AMAR
890 DATA HOUSE,CHATEAU,CASA
900 DATA MAN,HOMME,HOMBRE
910 DATA WOMAN,FEMME,HEMBRA
920 DATA I,JE,YO
930 DATA YOU,VOUS,USTED
940 DATA UNDERSTAND,COMPRENDRE,COMPRENDER
950 DATA PRETTY,JOLI,BONITO
960 DATA WHITE,BLANC,BLANCO
970 DATA STREET,RUE,CALLE
980 DATA ONE,UN,UNO
990 DATA TWO,DEUX,DOS
1000 DATA THREE,TROIS,TRES
1010 DATA RED,ROUGE,ROJO
1020 DATA END,END,END
1030 END

```

## How To Load These Cassette Programs Into Your TI-99/4A

This tape will load according to the OLD instructions in your *TI 99/4A User's Reference Guide* (the one that came with your computer). Briefly, here's what to do:

1. Connect and power up your TI-99/4A, a TV screen, and cassette recorder(s). Set the tape counter on CS1 to zero and insert the cassette. With your TI 99/4A in Command Mode, type **OLD CS1** and press ENTER.
2. Your screen will show a series of instructions, the first of which tells you to rewind the cassette. Instead of rewinding, proceed to the next step on this sheet.
3. Suppose the name of the program you want to load is **Siren**, which should be ready to load when 094 appears on the tape counter. Using the "Fast Forward" control on Recorder CS1, run the tape ahead until the counter indicates 094. Stop the tape. Press ENTER.
4. Now proceed with the next screen instruction, which is PRESS CASSETTE PLAY. Depress the "Play" button on the recorder. Then press ENTER on the computer.
5. Your screen will display READING as the program **Siren** is loaded and checked. On completion, you'll see DATA OK, followed by PRESS CASSETTE STOP . . . THEN PRESS ENTER. Depress the "Stop" button on the recorder. Then press ENTER on the computer. Your program is now loaded into memory, ready for running.

## If You Don't Know the Numbers Where Programs Start on the Tape

Just insert the cassette, fully rewound, and proceed as in Step 4 above. Your TI-99/4A will load the first program, as in Step 5. When you press the "Stop" button on the recorder, make a note of the tape counter number as this will be about where the next program starts. Continue loading each program, making a note of the counter numbers, and you'll end up with a complete program log.

## Programs on This Tape, With Locations

Fill the screen with your name . . . . .	002
What's your batting average? . . . . .	004
How far away is the lightning? . . . . .	007
If I double my investment every year, how much will it be worth? . . . . .	011
A routine to round off decimal numbers . . . . .	014
A program that can print backwards . . . . .	018
Changing from fractions to decimals . . . . .	021
A routine to right justify numbers . . . . .	025
A program to sort in alphabetical order . . . . .	029
Con-Cat-E-Nation . . . . .	034
Let the computer balance your checkbook . . . . .	039
Let the computer figure your family budget . . . . .	044
Turn your computer into a tv typewriter . . . . .	048
A program to help you understand random numbers . . . . .	052
A program to help you understand binary numbers . . . . .	057
A program to help you understand hexadecimal numbers . . . . .	062
Create and test your own special characters . . . . .	067
Creating a bar graph . . . . .	074
The Face . . . . .	079
Rocket . . . . .	086
Twinkle, Twinkle, Little Star . . . . .	090
Siren . . . . .	094
Joystick Crayon . . . . .	098
Keyboard Crayon . . . . .	103
Computer Piano . . . . .	108
Missile Control . . . . .	114
Name That Tone . . . . .	122
Trampoline . . . . .	129
Horse Race . . . . .	135
Bomb . . . . .	146
Remove the X's . . . . .	154
Repeat . . . . .	162
Scramble . . . . .	169
Sorting Batting Averages . . . . .	178
Dice . . . . .	186
Paper-Rock-Scissors . . . . .	193
Secret Code . . . . .	202
Micro Hangman . . . . .	209
Zodiac Signs . . . . .	220
Spelling Test . . . . .	227
Coin Count . . . . .	235
Checking Math Homework . . . . .	244
Guess the Missing Letters . . . . .	253
Savings Account . . . . .	263
Planets . . . . .	269
Temperatures . . . . .	278
Speed Reading . . . . .	287
State Capitals . . . . .	295
English to Metric . . . . .	310
Area of a Rectangle . . . . .	320
Language Translator . . . . .	328

## Notes


## Notes

## Notes









## **TI-99/4A: 51 Fun and Educational Programs**

Here are 51 programs that are good examples of TI BASIC in action. They are programs that have practical and useful applications. They are programs that you can change to fit your specific needs or they can be used just as they are.

- Work with numbers and characters and learn as you compute.
- Have fun playing games and working puzzles.
- Listen to your computer imitate a fire siren or sing "Twinkle, Twinkle, Little Star."
- Use graphics to draw pictures.
- Learn about the planets, the state capitals, or metric conversion, or learn foreign language words and their meanings.

If you are looking for programs that are *fun* and *educational*, you have found them.

**Howard W. Sams & Co., Inc.**  
4300 West 62nd Street, Indianapolis, Indiana 46268 U.S.A.